

05-590 Raszyn

```
; $ADDRESS=0000:07FF

; .XLIST
; TITLE CA80 MONITOR V3.01
; SUBTTL MIK08 Copyright (C)1987,1991 St.Gardynik

; Nowy CA80 na profesjonalnej płytce MIK290
; .Z80
; Oprogramowanie nowego CA80 na MIK290 25.08.1991
; Zmiany w CSTSM:,TKLAW:,NMI:,EMUL:,CA80:
; Układy Z80A CPU,8255,Z80A CTC opisano w MIK04
; Adresy portu systemowego typu 8255
; Strob PSYS - zobacz schemat ideowy (rys R6)
0000: PA EQU 0F0H
0000: PB EQU 0F1H
0000: PC EQU 0F2H
0000: CONTR EQU 0F3H
;
; Ustawienie konfiguracji dla portu systemowego
; KONF/90 - słowo sterujące
; PA - wejście, PB,PC - wyjście
;
0000: KONF EQU 90H
; Adresy portu emulatora typu 8255
; Strob EME8 - zobacz schemat ideowy (rys R8)
;
0000: PA1 EQU 0E8H
0000: PB1 EQU 0E9H
0000: PC1 EQU 0EAH
0000: CONTR1 EQU 0EBH
; Ustawienie konfiguracji dla portu emulatora
; PA - wejście /TRYB1/ PB - wyjście /TRYB1/
;
0000: KONF1 EQU 0B4H ;Słowo sterujące
; Adresy kanałów zegara typu Z80A CTC
; Strob CTF8 - zobacz schemat ideowy (rys R8)
;
0000: CHAN0 EQU 0F8H ;Kanał 0
0000: CHAN1 EQU 0F9H ;Kanał 1
0000: CHAN2 EQU 0FAH ;Kanał 2
0000: CHAN3 EQU 0FBH ;Kanał 3
; .COMMENT%
; Kanał 0 układu Z80A CTC
; Kanał nr. 0 jest zerowany za każdym
; razem kiedy przy pomocy zlecenia *C (praca kro-
; kowa) wykonany zostanie choćby jeden rozkaz
; użytkownika. Jeśli użytkownik nie korzysta
; z pracy krokowej to może wykorzystywać kanał 0
; do własnych celów.
; CCR0 - słowo sterujące dla kanału 0
; TC0 - stała dla timera
; %
0000: CCR0 EQU 87H ;Tryb timer
0000: TC0 EQU 10 ;Stała dla timera
0000: ZCHAN EQU 3 ;Stała zerująca kanał
; .COMMENT%
; Kanał 1 układu Z80A CTC
; Kanał 1 pracuje w trybie timer przy zablokowanych
; przerwaniach. Realizuje podział częstotliwości
```

wykorzystac

```
;zegara (f=4MHz) przez TC1*16=4000. Czesotliwosc
;na wyjsciu ZC/T01 wynosi zatem f=1kHz. Kanał
;inicjowany jest jednokrotnie natychmiast po
;wlaczeniu zasilania. Jesli wyjscie ZC/T01 nie
;jest wykorzystywane do generowania przerwan
;niemaskowalnych NMI to uzytkownik moze

;kanal 1 do wlasnych celow majac swiadomosc,
;ze jest on inicjowany j/w po wlaczeniu zasilania.
;CCR1 - slowo sterujace dla kanalu 1
;TC1 - stala dla timera
;%
0000: CCR1 EQU 7 ;Tryb timer
0000: TC1 EQU 250 ;Stala dla timera
;*** STALE SYMBOLICZNE ***
;.COMMENT%
;RESI - strob kasujacy zgloszenie przerwania
;maskowalnego jesli system przerwan mikroproce-
;sora ustawiony jest w tryb 1 (brak Z80A CTC).
;HLUZYT - inicjacja rejestrow HL uzytkownika
;PCUZYT - inicjacja rejestru PC uzytkownika
;WMSEK - wzorzec milisekund
;Warunek ktory musi byc spelniony dla potrzeb
;zegara czasu rzeczywistego
;
; FNMI/WMSEK=100 Hz gdzie:
;FNMI - czesotliwosc przerwan NMI
;FNMI=500 Hz - standart dla CA80
;GKLAW - kod tablicowy klawisza "G" (tablica
TKLAW)
;MKLA - kod rzeczywisty klawisza "M"
;%
0000: RESI EQU 0FCH ;Kasowanie INT
0000: SYGNAL EQU 0ECH ;Sygnal dzwiekowy
0000: HLUZYT EQU 0C100H ;HL- uzytkownika
0000: PCUZYT EQU 0C000H ;PC- uzytkownika
0000: WMSEK EQU 5 ;Wzorzec WMSEK
0000: GKLAW EQU 10H ;Kod tab. klaw. "G"
0000: SPAC EQU 11H ;Kod tab. klaw. "."
0000: CR EQU 12H ;Kod tab. klaw. "="
0000: MKLA EQU 58H ;Kod rzecz. klaw. "M"
0000: MKLA30 EQU MKLA&0FH ;Bity B3-B0
0000: MKLA64 EQU MKLA&70H ;Bity B6-B4
0000: KRP EQU 0F4D3H ;Kod rozkazu OUT
(0F4H),A
0000: RST30 EQU 0F7H ;Kod rozkaz. RST 30H
;
;Stale sterujace pojedyncza cyfra wyswietlacza
;- dla procedury COM
0000: GLIT EQU 3DH ;Kod siedmioseg. litery G
0000: ZGAS EQU 0 ;Zgaszenie cyfry wyswiet.
0000: KRESKA EQU 40H ;Zaswiec. srodkow. segment.
0000: ANUL EQU 8 ;Zaswiec. dolnego segmentu
0000: ROWN EQU 48H ;Znak rownosc
0000: KROP EQU 7 ;Zaswiecenie kropki
;
0000: ORG 0000H
;
0000: .LIST
```

```

; .COMMENT%
;Zmiany w programie glownym programu MONITORA 2KB
;dla nowego CA80 - na plytce MIK290.
;Oprogramowanie nowego CA80 (na MIK290) powyzej
;adresu 61DH jest IDENTYCZNE jak starego CA80 na
;plytce MIK90. Najwazniejsza zmiana dotyczy
;obslugi klawiatury i procedur STARTN:/261
;CSTSM:/130, ZNMI:/E4, EMUL:/5bd, TKLAW:/300
;***** P R O G R A M   G L O W N Y *****
;%
0000: 00      CA80:  NOP    ;Rozkazy NOP dodano, aby
0001: 00          NOP    ;ulatwic prawidlowy start
0002: 00          NOP    ;Z80 CPU przy wlaczaniu
0003: 00          NOP    ;zasilania
0004: C3 56 01    JP     CA80N ;Ciag dalszy programu
; .XLIST
; .COMMENT%
;TI - procedura systemowa
;Pobranie znaku z jednoczesnym jego wyswietleniem
;w/g PWYS. [ tzw.ECHO ]
;Wyswietlone zostana wytlacznie cyfry szesnastkowe
;- pozostale znaki beda pobrane lecz nie wyslane
;na wyswietlacz.
;
;WEJ: -
;WYJ:  A -   pobrany znak
;  CY=1  znak CR
;  Z=1 i CY=0 znak SPAC
;ZMIENIA: AF          STOS: 8
;
;Wywołanie:
;RST TI1  lub  CALL TI1  lub  CALL TI
;          DB    PWYS
;%
0007:      TI:
0007: EF      RST    USPWYS    ;Ustawienie PWYS
0008:      TI1:
0008: C5      PUSH   BC      ;Ochrona BC
0009: CD C6 FF    CALL   CI      ;Pobranie znaku
000C: F5      PUSH   AF      ;Ochrona AF
000D: 4F      LD     C,A
000E: 18 2B     JR     TI1cd ;Ciag dalszy
; .COMMENT%
;CLR - procedura systemowa
;Wygazzenie znakow wyswietlacza w/g parametru PWYS
;
;WEJ: PWYS - okresla ktore znaki maja byc
wygaszone
;WYJ: Odpowiednie znaki wygaszone
;ZMIENIA: AF          STOS: 4
;WYWOLANIE:
;RST CLR  lub  CALL CLR  lub  CALL CLR1
;DB PWYS    DB    PWYS
;%
0010:      CLR:
0010: EF      RST    USPWYS    ;Ustawienie PWYS
0011:      CLR1:
0011: C5      PUSH   BC
0012: 0E 00     LD     C,ZGAS    ;Wygazzenie cyfry

```

```

0014: 06 08          LD    B,8    ;Max. ilosc cyfr do wygasz
0016: 18 29          JR     CLR2   ;Ciag dalszy
; .COMMENT%
; LBYTE - procedura systemowa
; Wswietlenie rej. A w postaci dwucyfrowej liczby
; szesnastkowej w/g PWYS.
;
; WEJ: A- liczba do wswietlenia
; WYJ: wswietlenie liczby w/g PWYS
; ZMIENIA: F,C          STOS: 8
; WYWOLANIE:
; RST LBYTE lub CALL LBYTE lub CALL LBYTE1
; DB    PWYS          DB    PWYS
; Bity PWYS7-PWYS4 - dowolne
; Bity PWYS3-PWYS0 =<7,6,5...1,0> - nr. pozycji

wysw

; Jesli PWYS30=7 to wswietlenie tylko mniej
; znaczej cyfry na pozycji nr. 7
; (APWYS+1)(APWYS)=PWYS - inicjowane po wlaczeniu
; zasilania
;
; %
0018:          LBYTE:
0018: 4F          LD     C,A    ;Ochrona A
0019: EF          RST     USPWYS  ;Ustawienie PWYS
001A: 79          LD     A,C    ;Odtworzenie A
001B:          LBYTE1:
001B: E5          PUSH    HL
001C: D5          PUSH    DE    ;Ochrona HL i DE
001D: C3 0D 01    JP     LBYTcd  ;Ciag dalszy
; .COMMENT%
; LADR - procedura systemowa
; Wswietlenie HL w postaci czterocyfrowej liczby
; szesnastkowej w/g PWYS.
;
; WEJ: HL - liczba do wswietlenia
; WYJ: wswietlenie liczby w/g PWYS
; ZMIENIA: AF,C          STOS: 10
; WYWOLANIE:
; RST LADR lub CALL LADR lub CALL LADR1
; DB    PWYS          DB    PWYS
; Bity PWYS30=<7,6...1,0> - nr. pozycji

wswietlacza

; Bity PWYS74 - dowolne
; Jesli PWYS30>=5 to wswietlone zostana tylko

mniej

; znacze cyfry mieszczace sie w obrebie wswietl.
; (APWYS+1)(APWYS)=PWYS - inicjowane po wlaczeniu
; zasilania
;
; %
0020:          LADR:
0020: EF          RST     USPWYS  ;Ustawienie PWYS
0021:          LADR1:
0021: 7D          LD     A,L
0022: CD 1B 00    CALL    LBYTE1  ;Wysw. mlodszej bajtu
0025: 7C          LD     A,H
0026: 18 20          JR     LADRcd  ;Ciag dalszy
; .COMMENT%
; USPWYS - procedura pomocnicza
; Ustawienie parametru PWYS

```

```

;
;WEJ: SP+2 wskazuje mlodszy bajt adresu PCU
; - zobacz przyklad ponizej.
;ZMIENIA: A          STOS: 2
;
;PRZYKLAD:
;1. PROGRAM UZYTKOWNIKA
; CALL COM ;Wywołanie proc. COM
;PCU: DB 20H ;Parametr PWYS/20
;2. PROCEDURA SYSTEMOWA
;COM: RST USPWYS ;Wywołanie proc. USPWYS
;COM1:
;3. PROCEDURA POMOCNICZA USPWYS
;USPWYS: ;(STOS) = COM1,PCU !!
;Gdzie:
;PCU - adres komorki pamieci pod ktorym przecho-
;wywana jest nowa wartosc PWYS
;COM1 - adres powrotu do procedury COM
;
;Po wejsciu do procedury USPWYS na stosie przecho-
;wywane sa kolejno adresy COM1 i PCU !
;Procedura USPWYS powoduje:
;A. (PWYS):=(PCU) - przeslanie zawartosci komorki
;pamieci wskazywanej przez adres PCU do komorki
;pamieci wskazywanej przez adres PWYS
;Gdzie:
;(APWYS+1)(APWYS)=PWYS - adres PWYS przechowywany
;jest w komorkach pamieci wskazywanych przez adres
;APWYS (adresowanie posrednie).
;B. PCU:=PCU+1 - sztuczne zwiekszenie adresu
;powrotu do programu uzytkownika. Po zwiekszeniu
;PCU wskazuje kolejny rozkaz do wykonania
;w programie uzytkownika.
;%
0028: USPWYS:
0028: E5 PUSH HL
0029: D5 PUSH DE ;Ochrona HL,DE
;Na stosie schowane sa kolejno DE,HL,COM1,PCU
;SP wskazuje rejestr E
;SP+4 wskazuje mlodszy bajt adresu COM1
;SP+6 wskazuje mlodszy bajt adresu PCU
002A: 21 06 00 LD HL,6
002D: 39 ADD HL,SP ;HL- wskazuje PCU
002E: 18 25 JR USPWcd ;Ciag dalszy
;.COMMENT%
;RESTA - powrot do programu monitora.
;Wykonanie przez program uzytkownika rozkazu
;RST 30H/F7 spowoduje skok do procedury RESTAR
;gdyz: AREST: JP RESTAR
;%
0030: RESTA:
0030: F3 DI ;Maskowanie przerwan
0031: C3 C9 FF JP AREST ;Skok do RESTAR
0034: K02:
0034: 79 50 50 FF DB 79H,50H,50H,0FFH ;Komunikat "Err"
;.COMMENT%
;Skok do procedury obslugujacej przerwanie uzytko-
;wnika - pod warunkiem ze przerwania

```

mikroprocesora

```

;ustawione sa w tryb 1 (brak Z80A CTC).
;%
0038: C3 CF FF      JP      INTU  ;Obsluga przerwania INT
; .COMMENT%
;
;Dokonczenie procedur TI, CLR, LADR, USPWYS.
;
;%
;Dokonczenie procedury TI
TI1cd:
003B:              CALL  C01    ;Wysw. cyfry szesnastkow.
003B: CD E1 01      POP   AF     ;Odtworzenie AF
003E: F1            POP   BC     ;Odtworzenie BC
003F: C1            RET
0040: C9            ;Dokonczenie procedury CLR
CLR2:
0041:              CALL  COM1    ;Wygazzenie cyfry
0041: CD AC 01      DJNZ  CLR2
0044: 10 FB          POP   BC     ;Odtworzenie BC
0046: C1            RET
0047: C9            ;Dokonczenie procedury LADR
LADRcd:
0048:              PUSH  HL       ;Ochrona HL
0048: E5            LD    HL,(APWYS) ;HL- adres PWYS
0049: 2A C1 FF      ;Ustawienie PWYS30 dla potrzeb starszego bajtu
004C: 34            INC   (HL)
004D: 34            INC   (HL)
004E: CD 1B 00      CALL  LBYTE1    ;Wysw. starszego bajtu
;Odtworzenie PWYS30
0051: 35            DEC   (HL)
0052: 35            DEC   (HL)
0053: E1            POP   HL       ;Odtworzenie HL
0054: C9            RET
;Dokonczenie procedury PWYS
;
0055:              USPWcd:
;Pobranie PCU do rejestrow DE
0055: 5E            LD    E,(HL)     ;Mlodszy bajt PCU
0056: 23            INC   HL
0057: 56            LD    D,(HL)     ;Starszy bajt PCU
0058: 1A            LD    A,(DE)     ;Pobranie (PCU)
0059: 13            INC   DE         ;Zwiekszenie adresu PCU
;Odtworzenie PCU
005A: 72            LD    (HL),D    ;Starszy bajt
005B: 2B            DEC   HL
005C: 73            LD    (HL),E     ;Mlodszy bajt
005D: 2A C1 FF      LD    HL,(APWYS) ;Pobranie adresu PWYS
0060: 77            LD    (HL),A     ;(PWYS):=(PCU)
0061: D1            POP   DE
0062: E1            POP   HL       ;Odtworzenie HL,DE
0063: C9            RET
0064: 85            DB    85H      ;Rok powstania 1985
; .COMMENT%
;SPEC - procedura systemowa
;Powrot do programu wywołujacego
;WEJ: -
;WYJ: Powrot do programu wywołujacego
;ZMIENIA: -          STOS: 0
;%

```

```

0065:                                SPEC:
0065: C9                                RET
                                        ;.COMMENT%
                                        ;NMI - procedura obsługi przerwania

niemaskowalnego

                                        ;Obsługa klawiatury, zegara, wyświetlacza oraz
                                        ;badanie czy klawisz "M" jest wcisnięty.
                                        ;1.Jesli M wcisniety to inicjacja komorek pamieci
                                        ;RAM zawartych w obszarze <APWYS,NMIU>
                                        ;GSTAT - klucz programowy
                                        ;GSTAT=0 - wykonywany program uzytkownika
                                        ;GSTAT#0 - wykonywany program MONITORA
                                        ;2.Jesli M wcisniety i GSTAT=0 to zapamietanie
                                        ;stanu procesora uzytkownika i powrot do prog-
                                        ;ramu MONITORA.
                                        ;3.Jesli M wcisniety i GSTAT#0 to skok do adresu
                                        ;START - oczekiwanie na kolejne zlecenie.
                                        ;%
0066:                                NMI:
0066: F5                                PUSH AF
0067: E5                                PUSH HL
0068: D5                                PUSH DE
0069: C5                                PUSH BC ;Ochrona AF,HL,DE,BC
                                        ;.COMMENT%
                                        ;Obsługa klawiatury - generowanie sygnału wcis-
                                        ;nięcia klawisza.
                                        ;Liczniki LCI i SYG współpracują z procedura CI.
                                        ;%
006A: 21 E8 FF                        LD HL,LCI ;Adres licznika klawiat.
006D: AF                                XOR A ;Zerowanie A
006E: BE                                CP (HL) ;Czy LCI=0
006F: 28 01                            JR Z,KCI
0071: 35                                DEC (HL) ;Zmniejsz. licznika LCI
0072: 23                                KCI: INC HL ;Wskazuje licznik SYG
0073: BE                                CP (HL) ;Czy SYG=0?
0074: 28 03                            JR Z,KSYG ;Rowny zeru
0076: 35                                DEC (HL) ;Zmniejsz. licznika SYG
0077: D3 EC                            OUT (SYG),A ;Generowanie impulsu
0079: 23                                KSYG: INC HL ;Wskazuje licznik TIME
007A: 35                                DEC (HL) ;Zmniejsz. licznika TIME
007B: 3A B4 FF                        LD A,(ZESTAT)
007E: B7                                OR A ;Czy ZESTAT=0 ?
007F: 28 41                            JR Z,ZKON1 ;Zegar wyłączony
                                        ;Obsługa zegara czasu rzeczywistego
                                        ;
0081: 23                                INC HL ;Wskazuje MSEK
0082: 11 28 03                        LD DE,TABC ;Adres tablicy TABC
0085: 06 05                            LD B,LTABC ;Ilosc elementow w TABC
0087:                                PZEG:
0087: EB                                EX DE,HL
0088: 1A                                LD A,(DE) ;Pobranie czasu
0089: 3C                                INC A ;Zwiekszenie
008A: 27                                DAA ;W kodzie BCD
008B: BE                                CP (HL) ;Porownanie z ograniczeniem
008C: EB                                EX DE,HL ;HL - wskazuje czas
008D: 20 32                            JR NZ,ZKON
008F: AF                                XOR A ;Zerowanie A i CY
                                        ;CY=0 - wazne dla rozkazu DAA
0090: 77                                LD (HL),A

```

0091: 13	INC DE
0092: 23	INC HL
0093: 10 F2	DJNZ PZEG
	;DE - wskazuje TABM
0095: 35	DEC (HL) ;Zmniejsz. dni tygodnia
0096: 20 02	JR NZ,PZEG1
0098: 36 07	LD (HL),7 ;Dnityg <7,6,5..1>
009A:	PZEG1:
009A: 23	INC HL ;Wskaz. dni miesiaca
009B: 23	INC HL ;Wskazuje MIES
009C: 7E	LD A,(HL) ;Pobranie mies
	;Miesiace odliczane sa w kodzie BCD <1,12>
	;Zamiana kodu BCD na binarny
009D: FE 0A	CP 0AH
009F: 38 02	JR C,OKM ;Gdy MIES=<9
00A1: D6 06	SUB 6 ;Gdy MIES>9
	;Wyliczanie adresu w TABM
	;TABM musi lezec w obrebie strony
00A3:	OKM:
00A3: 3D	DEC A
00A4: 83	ADD A,E
00A5: 5F	LD E,A ;TABM musi byc na stronie
00A6: 1A	LD A,(DE) ;Pobranie ograniczenia
00A7: 57	LD D,A
00A8: 2B	DEC HL
00A9: 7E	LD A,(HL) ;Pobor. dni miesiaca
00AA: 3C	INC A
00AB: 27	DAA
00AC: BA	CP D ;Porownanie z ogranicz.
00AD: 38 12	JR C,ZKON
00AF: 3E 01	LD A,1
00B1: 77	LD (HL),A ;Inicjacja dni mies.
00B2: 23	INC HL
00B3: 7E	LD A,(HL) ;Pobranie miesiecy
00B4: 3C	INC A
00B5: 27	DAA
00B6: FE 13	CP 13H
00B8: 38 07	JR C,ZKON
00BA: 3E 01	LD A,1
00BC: 77	LD (HL),A ;Inicjacja miesiecy
00BD: 23	INC HL
00BE: 7E	LD A,(HL) ;Lata
00BF: 3C	INC A
00C0: 27	DAA
00C1: 77	ZKON: LD (HL),A
	;.COMMENT%
	;Obsluga wyswietlacza
	;Bufor wyswietlacza "BWYS" musi lezec w obrebie
	;strony!
	;MIK90 - dla potrzeb plytki MIK90 (U7=8255)
	;MIK94 - dla potrzeb plytki MIK94 (uklad zastepczy
	;ukladu 8255) - zobacz rys. R27 (MIK05B)
	;(SBUF)- bity B7,B6,B5 realizuja licznik binarny
	;modulo 8 sterujacy dekodern cyfr wyswiet.(74145)
	;%
00C2:	ZKON1:
00C2: 21 F5 FF	LD HL,SBUF ;MIK94
00C5: 7E	LD A,(HL)
00C6: C6 20	ADD A,20H ;Zwiekszenie licznika


```

00C8: 77          LD      (HL),A
00C9: 23          INC     HL
00CA: 23          INC     HL      ;Wskazuje BWYS
00CB: E6 E0      AND     0E0H   ;Wyciecie bitow licznika
00CD: 47          LD      B,A   ;Przechowanie stanu licznika
00CE: 3E FF      LD      A,0FFH
00D0: D3 F1      OUT     (PB),A   ;Wygazzenie wyswietl.
00D2: DB F2      IN      A,(PC)
00D4: E6 1F      AND     1FH   ;Zerow. starego licznika
00D6: B0          OR      B     ;Ustaw. nowej wartosci
00D7: 4F          LD      C,A   ;Nowa wart. portu PC
00D8: D3 F2      OUT     (PC),A   ;Wybranie kolejnej cyfry
00DA: 78          LD      A,B   ;Stan liczn. modulo 8
00DB: 07          RLCA
00DC: 07          RLCA
00DD: 07          RLCA      ;Licznik na bitach B2-B0
00DE: 85          ADD     A,L   ;Wylicz. adresu w BWYS
00DF: 6F          LD      L,A   ;BWYS w obrebie strony !
00E0: 7E          LD      A,(HL)   ;Pobranie znaku do wysw.
00E1: 2F          CPL
00E2: D3 F1      OUT     (PB),A   ;Wysw. znaku
00E4:          .LIST
; .COMMENT%
;Zmiana w procedurze NMI/66 - MIK08 str.1-6
;Badanie czy klawisz "M" jest wcisniety.
;Operacja badania nie moze zmienic aktualnego
;stanu portu wyjsciowego PC (MIK290)
;gdzy spowodowaloby to zaklocenia w procedurze

CSTS.

; %
00E4: 79          ZMNMI: LD      A,C   ;Aktualny stan portu PC
00E5: F6 0F      OR      0FH   ;Wymuszenie 1 na PC3-PC0
00E7: E6 FE      AND     0FEH   ;Wymuszenie 0 na PC0
00E9: D3 F2      OUT     (PC),A   ;Ustawienie PC dla M
00EB: DB F0      IN      A,(PA)   ;Odczyt matrycy klawiat.
00ED: 0F          RRCA      ;PA6-PA1 na A5-A0
00EE: E6 3F      AND     03FH   ;Zerowanie A7-A6
00F0: FE 3E      CP      3EH   ;Czy klawisz 'M'
00F2: 79          LD      A,C   ;Aktualny stan PC
00F3: D3 F2      OUT     (PC),A   ;Odtworzenie PC
00F5: CA 2F 05   JP      Z,MWCIS   ;Wcisniety klaw. 'M'
00F8: C1          POP     BC     ;Odtworzenie BC
00F9: CD CC FF   CALL    NMIU   ;Obsluga NMI uzytkownika
00FC: D1          POP     DE
00FD: E1          POP     HL
00FE: F1          POP     AF
00FF: ED 45      RETN
0101: FF FF FF FF DB      -1,-1,-1,-1
0105: FF FF FF FF DB      -1,-1,-1,-1
0109: FF FF FF FF DB      -1,-1,-1,-1
; .XLIST
;Dokonczenie procedury LBYTE
;
010D:          LBYTcd:
010D: 5F          LD      E,A   ;Ochrona A
010E: 2A C1 FF   LD      HL,(APWYS) ;Adres PWYS
0111: 7E          LD      A,(HL)   ;Pobranie PWYS
0112: 57          LD      D,A   ;Ochrona PWYS
0113: E6 0F      AND     0FH

```

```

0115: C6 10      ADD    A,10H ;PWYS74=1
0117: 77         LD     (HL),A      ;Wysw. bez przesuwania
0118: 7B         LD     A,E
0119: E6 0F      AND     0FH      ;Młodsza cyfra
011B: 4F         LD     C,A
011C: CD E1 01   CALL    C01      ;Wysw. młodszej cyfry
011F: 7B         LD     A,E
0120: 0F         RRCA
0121: 0F         RRCA
0122: 0F         RRCA
0123: 0F         RRCA
0124: E6 0F      AND     0FH      ;Starsza cyfra
0126: 4F         LD     C,A
0127: 34         INC     (HL)      ;PWYS-nast. pozycja
0128: CD E1 01   CALL    C01      ;Wysw. starszej cyfry
012B: 72         LD     (HL),D      ;Odtworzenie PWYS
012C: 7B         LD     A,E      ;Odtworzenie A
012D: D1         POP     DE
012E: E1         POP     HL      ;Odtworzenie HL i DE
012F: C9         RET

;
0130:          .LIST
; .COMMENT%  Zmiana w CSTSM - MIK08 str.1-9
; CSTS/FFC3 - procedura systemowa
;          Badanie czy klawisz wcisniety ?
; WEJ: -
; WYJ:  CY=1 - klawisz wcisniety
;      A   - kod tablicowy klawisza wcisniet.
;      ;
;      CY=0 - klawisz puszczoney
; ZMIENIA: AF          STOS: 2
; WYWOLANIE: CALL  CSTS      ;CSTS/FFC3
; %
; Realizacja skoku posredniego do CSTSM
; CSTS:  JP      CSTSM ;Wejscie do CSTSM !!!
; JP      CSTSM -inicjowane po wlaczeniu zasil.

0130:          CSTSM:
0130: E5          PUSH   HL
0131: C5          PUSH   BC
0132: 2E 04      LD      L,4      ;L-licznik
0134:          CST1:
0134: 2D          DEC     L      ;L=3,2,1,0,0FFH
0135: FA 6C 01   JP      M,CST2      ;Klaw. nie wcisniety
0138: 7D          LD      A,L      ;A=3,2,1,0
; Sterowanie bitami PC3-PC0 gdyz PC7-PC5
; nie moga ulec zmianie.
; PC75 - ustawiane w przerwaniu NMI!
0139: 07          RLCA      ;A0=0, A2-A1 nr linii w PC
013A: D3 F3      OUT     (CONTR),A ;Zerowanie linii w PC
013C: DB F0      IN      A,(PA) ;Odczyt matrycy klawiat.
013E: 0F         RRCA      ;Wazne PA6-PA1 na A5-A0
013F: E6 3F      AND     3FH      ;Tylko PA6-PA1
0141: 67         LD      H,A      ;Ochrona A
0142: 7D         LD      A,L      ;Aktualny nr. bitu
0143: 07         RLCA      ;A2-A1 nr. bitu
0144: 3C         INC     A      ;A0=1
0145: D3 F3      OUT     (CONTR),A ;Przywrocenie 1
0147: 7C         LD      A,H      ;Aktualny stan PA6-PA1
0148: FE 3F      CP      3FH      ;Czy klawisz wcisniety ?

```

```

014A: 28 E8      JR    Z,CST1      ;A=3FH,CY=0 nie wcisniety
                  ;Klawisz wcisniety H#3FH !
014C: 7D         LD    A,L      ;nr. bitu
014D: 0F         RRCA
014E: 0F         RRCA      ;Bity A1-A0 na A7-A6
014F: B4         OR     H        ;Suma logiczna z H
0150: C1         POP    BC      ;A-kod rzeczywisty klaw.
0151: E1         POP    HL      ;Odtworzenie BC,HL
0152: 18 09      JR     KONW    ;str.10 w MIK08
0154: FF FF      DB     -1,-1
                  ;Wlasciwy poczatek programu glownego
0156: 3E 90      CA80N: LD    A,KONF      ;Ustawienie konfigur.
0158: D3 F3      OUT    (CONTR),A ;PA-wejscie, PB,PC- wyj.
015A: C3 41 02   JP     CA80A ;Ciag dalszy
                  ; .XLIST
                  ;.COMMENT%
                  ;KONW - procedura pomocnicza
                  ;Konwersja kodu rzeczywistego klawisza na kod
                  ;tablicowy (tablica TKLAW).
                  ;WEJ: A - rzeczywisty kod klawisza
                  ;WYJ: CY=0 - klawisz nielegalny
                  ;      (nie istnieje w TKLAW)
                  ;      CY=1 - klaw. legalny (istnieje w TKLAW)
                  ;      A - kod tablicowy klawisza
                  ;ZMIENIA: AF          STOS: 2
                  ;%
015D: E5      KONW:  PUSH  HL
015E: C5      PUSH  BC      ;Ochrona HL i BC
015F: 21 00 03 LD    HL,TKLAW ;Adres tablicy TKLAW
0162: 06 18      LD    B,LTKLAW ;Dlugosc tablicy
0164:          CST5:
0164: BE      CP     (HL) ;Czy to ten ?
0165: 37      SCF      ;CY=1
0166: 28 04      JR     Z,CST2    ;Znaleziono !
0168: 23      INC    HL      ;Na nastepny kod rzeczyw.
0169: 10 F9      DJNZ   CST5     ;Szukaj dalej
016B: B7      OR     A        ;Klawisz nielegalny (CY=0)
016C: 7D      CST2: LD    A,L      ;Pobranie kodu tablicowego
016D: C1      POP    BC
016E: E1      POP    HL      ;Odtw. HL i BC
016F: C9      RET
                  ;*****
                  ;.COMMENT%
                  ;MA - zlecenie *A
                  ;Obliczanie sumy roznicy dwoch czterocyfrowych
                  ;liczb szesnastkowych.
                  ;*A[LICZBA1][SPAC][LICZBA2][CR]
                  ;%
0170: CD 13 02   MA:    CALL  EXPR ;Pobranie parametrow
0173: 40      DB     40H
0174: D1      POP    DE      ;LICZBA2
0175: E1      POP    HL      ;LICZBA1
0176: E5      PUSH   HL
0177: 19      ADD    HL,DE ;LICZBA1 + LICZBA2
0178: CD 20 00   CALL   LADR ;Wyswietlenie sumy
017B: 44      DB     44H
017C: E1      POP    HL
017D: B7      OR     A        ;CY=0
017E: ED 52      SBC    HL,DE ;LICZBA1 - LICZBA2

```

```

0180: CD 20 00      CALL LADR ;Wyswietlenie roznicy
0183: 40             DB 40H
;Wej. do proc. CIM - czekanie na wcis. klaw.
;*****
;.COMMENT%
;CI/FFC6 - procedura systemowa
;Pobranie znaku z klawiatury - czekanie dopoki
;klawisz nie zostanie puszczoney, a nastepnie wcis-
;niety. Rozpoznanie klawiszy CR,SPAC.
;
;WEJ: -
;WYJ: A - pobrany znak
; CY=1 - znak CR
; Z=1 i CY=0 - znak SPAC
;ZMIENIA: AF STOS: 4
;WYWOLANIE: CALL CI ;CI/FFC6
;%
;Realizacja skoku posredniego do adresu CIM
;CI: JP CIM ;Wej. do procedury CIM !
;JP CIM -inicjowane po wlaczeniu zasilania
CIM:
0184: E5            PUSH HL ;Ochrona HL
0185: 21 E8 FF      LD HL,LCI ;Licznik zmniejsz. w NMI
0188: 36 14          CI0: LD (HL),20 ;20*2=40 mS
018A: 7E            CI1: LD A,(HL)
018B: B7            OR A
018C: 20 FC          JR NZ,CI1 ;Opoznienie 40 mS
018E: CD C3 FF      CALL CSTS ;Czy klaw. wcisniety ?
0191: 38 F5          JR C,CI0 ;Czekaj na puszczenie
;Klawisz puszczoney
0193:                CI2:
0193: 36 14          LD (HL),20 ;40 mS
0195: 7E            CI3: LD A,(HL)
0196: B7            OR A
0197: 20 FC          JR NZ,CI3 ;Czekaj 40mS
0199: CD C3 FF      CALL CSTS ;Czy klawisz wcisniety ?
019C: 30 F5          JR NC,CI2 ;Czekaj na wcisniecie
;Klawisz wcisniety
019E: 23            INC HL ;Wskazuje licznik SYG
019F: 36 32          LD (HL),50 ;50*2=100mS
;Sygnal dzwiekowy generow. jest przez 100 mS
;Realizuje przerwanie NMI
01A1: E1            POP HL ;Odtworzenie HL
;.COMMENT%
;CRSPAC - procedura pomocnicza
;Badanie czy znak w rej. A jest CR lub SPAC
;
;WEJ: A - tablicowy kod znaku (tablica TKLAW)
;WYJ: CY=1 - znak CR
; Z=1 i CY=0 - znak SPAC
; Z=0 - znak inny niz CR lub SPAC
;ZMIENIA: F STOS: 0
;%
01A2:                CRSPAC:
01A2: FE 11          CP SPAC
01A4: C8            RET Z ;Z=1 i CY=0 - SPAC
01A5: FE 12          CP CR
01A7: 37            SCF
01A8: C8            RET Z ;CY=1 - znak CR

```

```

01A9: 3F          CCF
01AA: C9          RET          ;Z=0 - inny niz CR,SPAC
; .COMMENT%
; COM - procedura systemowa
; Wswietlenie znaku ktorego kod siedmiosegmentowy
; umieszczony jest w rejestrze C wedlug PWYS.
;
; WEJ: C - znak do wswietlenia
; WYJ: wswietlenie znaku w/g PWYS
; ZMIENIA: AF          STOS: 3 COM
;          STOS: 2 COM1
; WYWOLANIE:
; CALL COM lub CALL COM1
; DB PWYS
; Jesli PWYS nielegalne to natychmiastowy
; powrot bez wswietlenia.
; PWYS30=<7,6,5...0> - legalne
; PWYS74=<8,7,6...1> - legalne
; (APWYS+1)(APWYS)=PWYS - inicjowane po wlaczeniu
; zasilania.
; Bufor BWYS musi lezec w obrebie strony !
; %
01AB:          COM:
01AB: EF          RST USPWYS          ;ustawienie PWYS
01AC:          COM1:
01AC: E5          PUSH HL
01AD: C5          PUSH BC          ;Ochrona HL i BC
01AE: 2A C1 FF    LD HL,(APWYS) ;Adres PWYS
01B1: 4E          LD C,(HL)          ;Pobranie PWYS
01B2: 79          LD A,C
01B3: 0F          RRCA
01B4: 0F          RRCA
01B5: 0F          RRCA
01B6: 0F          RRCA
01B7: E6 0F      AND 0FH          ;Ilosc znakow angazow.
01B9: 47          LD B,A
01BA: 28 35      JR Z,C02 ;0 znakow angazowanych
01BC: 79          LD A,C
01BD: E6 0F      AND 0FH          ;Nr. pozycji wswiet.
01BF: 80          ADD A,B
01C0: FE 09      CP 9
01C2: 30 2D      JR NC,C02          ;Nielegalne PWYS
01C4: 85          ADD A,L
01C5: 6F          LD L,A
; HL - adres najstarszej angazowanej cyfry
; Bufor BWYS musi lezec w obrebie strony
01C6:          COM2:
01C6: 05          DEC B
01C7: 28 07      JR Z,COM3          ;Wszyst. znaki przesun.
01C9: 2B          DEC HL
01CA: 7E          LD A,(HL)
01CB: 23          INC HL
01CC: 77          LD (HL),A          ;Przes. znaku w BWYS
01CD: 2B          DEC HL
01CE: 18 F6      JR COM2 ;Przes. nastepny znak
01D0:          COM3:
01D0: C1          POP BC
01D1: 71          LD (HL),C          ;Wswietlenie znaku
01D2: E1          POP HL

```

```

01D3: C9          RET
                  ;.COMMENT%
                  ;PRINT - procedura systemowa
                  ;Wyswietlenie komunikatu w/g PWYS
                  ;WEJ: HL - adres pierwszego znaku do wyswietlenia.
                  ;      Po ostatnim znaku komunikatu musi byc
                  ;      0FFH - kryterium konca.
                  ;WYJ: wyswietlenie komunikatu w/g PWYS
                  ;ZMIENIA: AF,HL,C          STOS: 3
                  ;WYWOLANIE:
                  ;CALL PRINT lub CALL PRINT1
                  ;DB PWYS
                  ;%
01D4:          PRINT:
01D4: EF          RST USPWYS
01D5:          PRINT1:
01D5: 7E          LD A,(HL)          ;Pobranie znaku
01D6: FE FF          CP 0FFH          ;Czy znak 0FFH ?
01D8: C8          RET Z          ;Wroc gdy 0FFH
01D9: 4F          LD C,A
01DA: CD AC 01      CALL COM1          ;Wyswietlenie znaku
01DD: 23          INC HL          ;Nastepny znak
01DE: 18 F5          JR PRINT1
                  ;.COMMENT%
                  ;CO - procedura systemowa
                  ;Wyswietlenie cyfry szesnastkowej umieszczonej
                  ;w rej. C w/g PWYS
                  ;
                  ;WEJ: C - cyfra do wyswietlenia
                  ;      C=<0FH - cyfra legalna
                  ;WYJ: wyswietlenie cyfry w/g PWYS
                  ;ZMIENIA: AF          STOS: 5
                  ;WYWOLANIE:
                  ;CALL CO lub CALL C01
                  ;DB PWYS
                  ;Jesli cyfra nielegalna to natychmiast. powrot
                  ;TSIED - tablica w ktorej przechowywane sa kody
                  ;siedmiosegmentowe wszystkich cyfr szesnastkow.
                  ;%
01E0:          CO:
01E0: EF          RST USPWYS          ;Ustawienie PWYS
01E1:          CO1:
01E1: E5          PUSH HL
01E2: C5          PUSH BC          ;Ochrona HL i BC
01E3: 21 18 03      LD HL,TSIED ;Adr. tablicy TSIED
01E6: 79          LD A,C
01E7: FE 10          CP 10H
01E9: 30 06          JR NC,C02          ;Cyfra nielegalna
01EB: 85          ADD A,L          ;TSIED- w obrebie strony !
01EC: 6F          LD L,A
01ED: 4E          LD C,(HL)          ;Pobranie kodu cyfry
01EE: CD AC 01      CALL COM1          ;Wyswietlenie
01F1: C1          CO2: POP BC
01F2: E1          POP HL
01F3: C9          RET
                  ;.COMMENT%
                  ;PARAM - procedura systemowa
                  ;Pobieranie do rejestrow HL czterocyfrowej
                  ;liczby szesnastkowej z jednoczesnym jej

```

```

;wyswietlaniem w/g PWYS
;
;WEJ: wciskanie klawiszy bedacych cyframi sze-
;snastkowymi dopoty, dopoki CR lub SPAC.
;Znaki falszywe sa ignorowane zas pierwszy
;znak musi byc rozny od CR lub SPAC.
;
;WYJ: HL - pobrana liczba szesnastkowa. Jesli
;      wcisnieto wiecej niz 4 cyfry to wazne
;      sa ostatnie 4.
;      CY=1 ostatni wcisniety znak byl CR
;      CY=0 ostatni wcisniety znak byl SPAC
;      Wygaszenie wyswietlacza w/g aktualnego PWYS
;ZMIENIA: AF,HL          STOS: 9
;WYWOLANIE:
;CALL  PARAM lub  CALL  PARAM1
;DB    PWYS
;/%
01F4:  PARAM:
01F4:  RST    USPWYS      ;Ustawienie PWYS
01F5:  PARAM1:
01F5:  RST    TI1      ;Pobranie pierwsz. znaku
01F6:  JR     Z,PARAM1 ;Gdy CR lub SPAC
;
;.COMMENT%
;PARA1 - procedura systemowa
;Dzialanie identyczne jak PARAM1 lecz pierwsza
;cyfra szesnastkowa dostarczona w rej. A.
;-----
;Musi byc spelnione A=<0FH
;WYWOLANIE: CALL  PARA1
;/%
01F8:  PARA1:
01F8:  LD     HL,0
01FB:  PAR1:  PUSH  AF      ;Ochrona AF
01FC:  CP     10H      ;Czy cyfra szesnast. ?
01FE:  JR     NC,PAR2   ;Nie szesnastkowa
0200:  POP    AF      ;Odtw. wskaznikow
0201:  ADD    HL,HL
0202:  ADD    HL,HL
0203:  ADD    HL,HL
0204:  ADD    HL,HL ;Przes. w lewo o 4 bity
0205:  OR     L      ;Dopisanie ostat. cyfry
0206:  LD     L,A
0207:  PAR3:  RST    TI1   ;Pobor. nast. znaku
0208:  JR     PAR1
020A:  PAR2:  POP    AF      ;Odtw. znaku i wskaznik.
020B:  JR     NZ,PAR3   ;Inny niz CR lub SPAC
020D:  PUSH  AF      ;Ochrona AF
020E:  CALL  CLR1   ;Zgasz. angazowanych cyfr
0211:  POP    AF      ;Odtworzenie AF
0212:  RET
;
;.COMMENT%
;EXPR - procedura systemowa
;Pobranie ciagu czterocyfrowych liczb szesnastko-
;wych z jednoczesnym ich wyswietlaniem.
;
;WEJ: C - ilosc parametrow (liczb) do pobrania
;WYJ: pobrany ciag parametrow umieszcz. na stosie
;ZMIENIA: AF,HL,C          STOS: 10

```

```

;
;WYWOLANIE:
;CALL  EXPR  lub  CALL  EXPR1
;DB      PWYS
;[LICZBA1][SPAC][LICZBA2][SPAC]...[LICZBAn][CR]
;
;Klawisz SPAC sygnalizuje koniec pobierania
;liczby bieżącej i początek pobierania liczby
;następnej.
;Ostatni klawisz musi być CR.
;jeśli omyłkowo zamiast CR wcisniemy SPAC lub
;odwrotnie to nastąpi wyświetlenie komunikatu
;o błędzie lokalnym - znak "-". Ostatnio wcis-
;kana liczbę należy wówczas wprowadzić ponownie.
;Pobierane liczby chomikowane są kolejno na stosie
;Długość ciągu określona jest w rej. C.
;%
0213:      EXPR:
0213: EF      RST    USPWYS      ;Ustawienie PWYS
0214:      EXPR1:
0214: CD F5 01  CALL  PARAM1      ;Pobranie liczby
0217: E3      EX      (SP),HL
;Liczba należy schować przed adresem powrotu
0218: E5      PUSH   HL      ;Chow. adr. powrotu
0219: 0D      DEC     C
021A: 28 0E    JR      Z,EXP2      ;Koniec pobier. ciągu
021C: 30 F6    JR      NC,EXPR1 ;Pobierz nast. liczbę
;Obsługa błędu lokalnego
021E:      EXP1:
021E: C5      PUSH   BC      ;Ochrona BC
021F: 0E 08    LD      C,ANUL      ;Znak anulowania
0221: CD AC 01  CALL  COM1      ;Wysw. znaku anulow.
0224: C1      POP     BC      ;Odtw. BC
;Kasowanie ostatnio pobranej liczby
0225: E1      POP     HL      ;Adr. powrotu
0226: E3      EX      (SP),HL      ;Kasowanie
0227: 0C      INC     C      ;Przywr. stanu rej. C
0228: 18 EA    JR      EXPR1 ;Probuje raz jeszcze
022A: D8      EXP2:  RET     C      ;Ostatni musi być CR
022B: 18 F1    JR      EXP1 ;Wcisnięto SPAC
;.COMMENT%
;CZAS - procedura systemowa
;Wyświetlenie aktualnego czasu lub daty
;
;WEJ: HL=SEK/FFED - wyświetlenie czasu
;      HL=DNITYG/FFF0 - wyświetlenie daty
;WYJ: wyświetlenie aktualnego czasu lub daty
;ZMIENIA: AF,C      STOS:9
;WYWOLANIE:  CALL  CZAS
;%
022D:      CZAS:
022D: 7E      LD      A,(HL)
022E: DF      RST     LBYTE ;SEK
022F: 20      DB      20H
0230: 23      INC     HL
0231: 7E      LD      A,(HL)
0232: DF      RST     LBYTE ;MIN
0233: 23      DB      23H
0234: 23      INC     HL

```



```

0235: 7E          LD      A,(HL)
0236: DF          RST      LBYTE ;GODZ
0237: 26          DB       26H
0238: 2B          DEC      HL
0239: 2B          DEC      HL      ;Odtworzenie HL
023A: C9          RET

; .COMMENT%
; HILO - procedura systemowa
; Zmniejszenie rej. HL o 1 a nastepnie testowanie
; czy DE>=HL
; WEJ: HL,DE parametry wejscowe
; WYJ: HL=HL+1 , DE-HL
; Jesli CY=0 to DE>=HL
; Jesli CY=1 to DE< HL
; ZMIENIA: AF,HL          STOS: 0
; WYWOLANIE: CALL HILO
; %

023B:          HILO:
023B: 23          INC      HL
023C: 7B          LD      A,E      ;DE-HL
023D: 95          SUB      L
023E: 7A          LD      A,D
023F: 9C          SBC      A,H
0240: C9          RET

; C. D. - P R O G R A M U   G L O W N E G O
;

0241:          CA80A:
0241: 31 8D FF      LD      SP,TOS      ;Ustaw. stosu systemow.
; Inicjacja obszaru RAM angazowanego przez
; program MONITORA. Tablica TRAM/5C8
0244: 21 02 06      LD      HL,KTRAM
0247: 11 D1 FF      LD      DE,INTU+2 ;Pkt. 5.0 MIK05
024A: 01 3B 00      LD      BC,LTRAM ;Dlugosc bloku
024D: ED B8          LDDR
024F: 3E FF          LD      A,TOS/256 ;Starszy bajt TOS/FF8D
0251: ED 47          LD      I,A      ;Inicjacja rej. I
0253: ED 56          IM      1      ;Przerwania "TRYB 1"
; Inicjacja ukladu zegara Z80A CTC
; LOW INTU0 - mlodszy bajt adresu INTU0/FFD0
0255: 3E D0          LD      A,INTU0&255 ;Wektor dla Z80A CTC
0257: D3 F8          OUT     (CHAN0),A ;Wpisanie wektora
; Ustawienie kanalu 1 ukladu Z80A CTC w tryb
; "TIMER". Czyst. wyjsciowa ZC/T01=1 kHz
0259: 3E 07          LD      A,CCR1      ;Slovo sterujace
025B: D3 F9          OUT     (CHAN1),A ;Wpisanie CCR1
025D: 3E FA          LD      A,TC1      ;Stala dla timera
025F: D3 F9          OUT     (CHAN1),A ;Wpisanie stalej TC1
0261:          .LIST
; Zmiana tuz przed wejsciem do petli glownej
; MONITORA (2KB) - MIK08 str.1-17
; Jesli PA0=1 (8255) to skok do emulatora MSID
0261: ED 5E          STARTN: IM      2      ;Przerwania w tryb 2
0263: DB F0          IN      A,(PA)      ;Odczyt portu PA
0265: 0F          RRCA      ;CY=bit A0
0266: DA BE FF      JP      C,RTS ;Skok do emulatora
0269: 18 05          JR      START ;Do petli glownej
026B: FF FF FF FF      DB      -1,-1,-1,-1
026F: FF          DB      -1
; .XLIST

```

```

; .COMMENT%
; START - oczekiwanie na wprow. nowego zlecenia
;       Wejscie glowne do programu MONITORA
;
; Wszystkie zlecenia progr. MONITORA koncza sie
; skokiem do etykiety "START". Nacisniecie w do-
; wolnej chwili klaw. "M" rowniez konczy sie
; skokiem do "START"
; %
; Petla glowna programu MONITORA
0270: START:
0270: 31 8D FF      LD    SP,TOS      ;Ust. stosu systemowego
0273: D7           RST    CLR      ;Zerowanie wyswietlacza
0274: 80           DB     80H      ;Wszystkie cyfry
0275: 21 39 03     START1: LD    HL,K01    ;Poczatek komunikatu
0278: CD D4 01     CALL   PRINT ;Wyswiet. CA80
027B: 40           DB     40H
027C: CD BD 05     CALL   EMUL  ;Sprawdz. czy emulat.
027F: CD 07 00     CALL   TI     ;Pobranie zlecenia
0282: 17           DB     17H      ;Najstarsza cyfra wysw.
0283: 5F           LD     E,A      ;Numer zlecenia
0284: FE 11        CP     LCT      ;Czy zlec. legalne ?
0286: F2 87 04     JP     P,ERROR ;Nielegalne !
0289: FE 10        CP     GKLAW    ;Czy zlec. G
028B: 20 06        JR     NZ,INNE   ;Inne niz G
028D: 0E 3D        LD     C,GLIT    ;Kod siedmioseg. G
028F: CD AB 01     CALL   COM      ;Wyswietlenie "G"
0292: 17           DB     17H      ;PWYS
0293: D7           INNE:  RST    CLR      ;Kasuj "rr" z kom. "Err"
0294: 70           DB     70H
0295: 01 70 02     LD     BC,START ;Adres powrotu
0298: C5           PUSH   BC      ;Na stos
0299: 0E 02        LD     C,2      ;2 parametry dla EXPR
; Wyliczenie adresu pod ktorym przechowywany
; jest adres procedury obslugujacej zlecenie.
029B: 21 A7 02     LD     HL,CTBL    ;Tablica zleceń
029E: 16 00        LD     D,0      ;E- nr. zlecenia !
02A0: 19           ADD     HL,DE
02A1: 19           ADD     HL,DE ;HL=HL+2*DE
; Pobranie adresu procedury
02A2: 5E          LD     E,(HL)
02A3: 23          INC     HL
02A4: 56          LD     D,(HL)
02A5: EB          EX     DE,HL
02A6: E9          JP     (HL) ;Pseudo CALL do proced.
; Tablica zleceń
; M0 - zlecenie nr. 0 (klawisz nr.0) itd
02A7: CTBL:
02A7: C9 02        DW     M0      ;Wyswietlenie zegara
02A9: DC 02        DW     M1      ;Ustawienie czasu
02AB: ED 02        DW     M2      ;Ustawienie daty
02AD: B4 04        DW     M3      ;Wymiana rej. procesora
02AF: 1D 06        DW     M4      ;Zapis na magnetofon
02B1: 74 06        DW     M5      ;Zapis rekordu EOF
02B3: 14 07        DW     M6      ;Odczyt z magnetofonu
02B5: CB 04        DW     M7      ;Parametry transmisji
; Inicjacja CA80
02B7: B5 FF        DW     M8      ;Zlecenie uzytkownika
02B9: DE 04        DW     M9      ;Poszuk. slowa 8-16 bit

```

```

02BB: 70 01      DW  MA      ;Suma i roznica hex.
02BD: FF 04      DW  MB      ;Przesun. obszaru PAM.
02BF: 3E 03      DW  MC      ;Praca krokowa
02C1: 72 03      DW  MD      ;Przeglądanie pamieci
02C3: 97 03      DW  ME      ;Wpisanie stalej
02C5: AE 03      DW  MF      ;Przeglądanie rejestr.
02C7: 66 04      DW  MG      ;Skok do progr. uzytkow.
02C9:            LCT  EQU    ($-CTBL)/2
; .COMMENT%
; *****
; * Zlecenia programu MONITORA *
; *****
; %
; M0 - wyswietlenie zegara GODZ/MIN/SEK
;
02C9:            M0:
02C9: 21 ED FF      LD    HL,SEK      ;Wysw. czasu
02CC: CD 2D 02      CALL  CZAS      ;GODZ/MIN/SEK
02CF: CD C3 FF      M01: CALL  CSTS      ;Czy klawisz wcisnien.?
02D2: 30 F5          JR    NC,M0      ;Nie wcisniety
02D4: 21 F1 FF      LD    HL,DNIM      ;Klawisz wcisniety
02D7: CD 2D 02      CALL  CZAS      ;ROK/MIES/DZIEN
02DA: 18 F3          JR    M01
; M1 - ustawienie czasu
; *1[GODZ][SPAC][MIN][SPAC][SEK][CR]
; ZMIENIA: AF,BC,HL      STOS: 11
;
02DC: 0C            M1:    INC    C      ;3 parametry
02DD: CD 13 02      CALL  EXPR      ;Pobranie parametrow
02E0: 20            DB    20H      ;PWYS
02E1: 21 ED FF      LD    HL,SEK      ;Adres SEK
02E4: C1            DATUST: POP  BC
02E5: 71            LD    (HL),C      ;SEK
02E6: 23            INC    HL
02E7: C1            POP  BC
02E8: 71            LD    (HL),C      ;MIN
02E9: C1            POP  BC
02EA: 23            INC    HL
02EB: 71            LD    (HL),C      ;GODZ
02EC: C9            RET
; .COMMENT%
; M2 - ustawienie ROK/MIES/DZIEN MIESIACA/DZIEN TYG
; *2[ROK][SPAC][MIES][SPAC][DNIM][SPAC][DNITYG][CR]
; ZMIENIA: AF,BC,HL      STOS: 11
; %
02ED: 0E 04          M2:    LD    C,4      ;4 parametry
02EF: CD 13 02      CALL  EXPR      ;Pobranie parametrow
02F2: 20            DB    20H
02F3: 21 F0 FF      LD    HL,DNITYG
02F6: C1            POP  BC
02F7: 71            LD    (HL),C      ;DNI TYG.
02F8: 23            INC    HL
02F9: 18 E9          JR    DATUST
; *****
02FB: CD 94 03      ZMD:    CALL  SU1      ;Zlecenie *D
02FE: 18 3F          JR    MC1      ;c.d. zlecenia MC !!
; *****
0300:            .LIST
; .COMMENT%      Zmiany w TKLAW - MIK08 str.1-19

```

```

;TKLAW - tablica klawiatury
;Zawiera kod rzeczywisty kazdego klawisza i od-
;powiadajacy mu kod tablicowy.
;Kod tablicowy - to mniej znaczacy bajt adresu
;wskazujacego kod rzeczywisty. Stad:
;TKLAW - musi rozpoczynac sie od poczatku
;strony !!. Kod rzeczywisty klawisza pobierany
;jest przez procedure CSTS a nastepnie prze-
;twarzany na kod tablicowy w procedurze KONW.
;Objasnienia komentarzy:
;G/10 - klawisz "G" ma kod tablicowy 10H
; (kod rzeczywisty - 7EH)
;%
0300: TKLAW:
0300: FB DB 0FBH ;0/0
0301: EF DB 0EFH ;1/1
0302: FD DB 0FDH ;2/2
0303: DF DB 0DFH ;3/3
0304: BB DB 0BBH ;4/4
0305: AF DB 0AFH ;5/5
0306: BD DB 0BDH ;6/6
0307: 9F DB 09FH ;7/7
0308: 7B DB 07BH ;8/8
0309: 6F DB 06FH ;9/9
030A: 7D DB 07DH ;A/0AH
030B: 5F DB 05FH ;B/0BH
030C: 3B DB 03BH ;C/0CH
030D: 2F DB 02FH ;D/0DH
030E: 3D DB 03DH ;E/0EH
030F: 1F DB 01FH ;F/0FH
0310: 7E DB 07EH ;G/10H
0311: BE DB 0BEH ;SPAC/11H
0312: FE DB 0FEH ;CR/12H
0313: 3E DB 03EH ;M/13H
0314: F7 DB 0F7H ;W=F4/14H
0315: B7 DB 0B7H ;X=F3/15H
0316: 77 DB 077H ;Y=F2/16H
0317: 37 DB 037H ;Z=F1/17H
0318: LTKLAW EQU $-TKLAW
;W,X,Y,Z - stary CA80 na MIK90
;F1,F2,F3,F4 - nowy CA80 na MIK290
; .XLIST
;.COMMENT%
;TSIED - tablica zawierajaca kody siedmiosegmen-
;towe cyfr szesnastkowych dla potrzeb wyswietl.
;Przyklad:
;Budowa kodu dla cyfry 0.
;K G F E D C B A
;0 0 1 1 1 1 1 1 = 3FH
;Segment srodkowy "G" oraz kropka "K" musza byc
;wygaszone. Swieca sie segmenty: A,B,C,D,E,F
;TSIED - musi lezec w obrebie strony !!!
;%
0318: TSIED:
0318: 3F 06 5B 4F DB 3FH,6,5BH,4FH ;0,1,2,3
031C: 66 6D 7D 07 DB 66H,6DH,7DH,7 ;4,5,6,7
0320: 7F 6F 77 7C DB 7FH,6FH,77H,7CH ;8,9,A,B
0324: 39 5E 79 71 DB 39H,5EH,79H,71H ;C,D,E,F
;.COMMENT%

```

```

;TABC - tablica ograniczen czasowych
;TABM - tablica ograniczen miesiecy
;TABC,TABM - dla potrzeb zegara czasu rzeczy-
;wistego realizowanego w procedurze MNI.
;TABM - musi lezec w obrebie strony i musi
;byc umieszczona bezposrednio pod TABC.
;%
0328:      TABC:
0328: 05      DB      WMSEK ;Wzorzec milisekund
0329: 00      DB      0      ;SETNE SEK
032A: 60      DB      60H    ;Sekundy
032B: 60      DB      60H    ;MIN
032C: 24      DB      24H    ;GODZ
032D:      LTABC EQU      $-TABC
;TABM musi byc pod TABC !!!
032D:      TABM:
032D: 32      DB      32H    ;Styczen
032E: 29      DB      29H    ;Luty
032F: 32      DB      32H    ;Marzec
0330: 31      DB      31H    ;Kwiecien
0331: 32      DB      32H    ;Maj
0332: 31      DB      31H    ;Czerwiec
0333: 32      DB      32H    ;Lipiec
0334: 32      DB      32H    ;Sierpien
0335: 31      DB      31H    ;Wrzesien
0336: 32      DB      32H    ;Pazdziernik
0337: 31      DB      31H    ;Listopad
0338: 32      DB      32H    ;Grudzien
; .COMMENT%
;Kod siedmiosegmentowy komunikatu powitalnego
;      "CA80"
;%
0339: 39 77 7F 3F FF      K01:      DB      39H,77H,7FH,3FH,0FFH ;CA80
;
; .COMMENT%
;MC - realizacja pracy krokowej
;[CR] - naciśnięcie klawisz CR spowoduje skok
;do programu użytkownika, wykonanie 1 rozkazu
;i powrót do procedury MC.
;[SPAC] - naciśnięcie klawisza SPAC spowoduje
;przejsie do procedury MD z mozliwoscia bezp-
;osred. powrotu do MC (naciśnięcie klaw. G).
;Naciśnięcie nazwy ktoregokolwiek z rejestrow
;spowoduje przejście do procedury MF z mozliw.
;bezposredniego powrotu do MC (naciśnięcie CR)
;
;*C - wyswietlenie zawartosci PC użytkownika
;i komorki pamieci wskazywanej przez PC - roz-
;kaz do wykonania.
;
;Przykładowy algorytm korzyst. ze zlecenia MC:
;1.[CR]...[CR] - wykonywanie programu użytkow-
;wnika rozkaz po rozkazie.
;2.[NAZWA REJESTRU] - przejście do procedury MF
; (przeglądanie i modyfikacja rej. procesora)
;3.[CR] - powrót do procedury MC
;4.[SPAC] - przejście do procedury MD.
; (przeglądanie i modyfikacja komorek pamieci)
;5.[G] - powrót do procedury MC

```

```

;6.[CR]...[CR] - jak w pkt. 1
;%
033E: F1      MC:      POP      AF      ;Zlikwidow. adr. powr.
033F: 2A A9 FF MC1:      LD      HL,(PLOC-1) ;PC- uzytkownika
0342: D7      RST      CLR      ;Zerow. wyswietlacza
0343: 70      DB      70H      ;7 mlodszych cyfr
0344: E7      RST      LADR     ;Wyswietlenie PC
0345: 43      DB      43H
0346: 7E      LD      A,(HL)     ;Pobranie (PC)
0347: DF      RST      LBYTE    ;Wyswietlenie (PC)
0348: 20      DB      20H
0349: CF      RST      TI1      ;Czekaj na wcisn. klaw.
034A: 20 21    JR      NZ,ZMF     ;Inny niz CR lub SPAC
034C: 30 AD    JR      NC,ZMD     ;Klawisz SPAC
; .COMMENT%
;Wcisnieto klawiisz CR
;Wejscie do programu uzytkownika z wymuszeniem
;przerwania po wykonaniu jednego rozkazu. Po-
;wrot poprzez procedure "RESTAR" do adresu MC1.
;KRP/F4D3 - zostanie umieszcz. w pam. jak nizej
;KROK: D3F4 ;OUT (0F4H),A
;%
034E: 21 D3 F4 LD      HL,KRP      ;KRP- OUT (0F4H),A
0351: 22 A2 FF LD      (KROK),HL
; .COMMENT%
;RESTAR - powrot z programu uzytkownika do
;programu MONITORA.
;Podpiecie procedury RESTAR do systemu przerwan.
;%
0354: 21 46 05 LD      HL,RESTAR
0357: 22 D0 FF LD      (INTU0),HL ;Inicjacja
;Synchronizacja z przerwaniem NMI
;TIME - licz. binarny zmniejszany w kazdym NMI
035A: 21 EA FF LD      HL,TIME
035D: 7E      LD      A,(HL)
035E: BE      SYN:      CP      (HL)
035F: 28 FD    JR      Z,SYN ;Synchronizacja
;Inicjacja kanalu nr. 0 ukkladu Z80A CTC
;Kanal zglosi przerwanie po TC0*16=160 taktach
;zegara. Musi to nastapic w trekcie wykonywania
;pierwszego rozkazu uzytkownika.
0361: 3E 87    LD      A,CCR0     ;Slovo sterujace
0363: D3 F8    OUT      (CHAN0),A ;Tryb "TIMER"
0365: 3E 0A    LD      A,TC0     ;Stala TC0
0367: D3 F8    OUT      (CHAN0),A ;Przerw. po 160 takt.
0369: 00      NOP      ;Dolozenie 4 taktow
036A: C3 AB 04 JP      G05      ;Do zlecenia *G
036D:          ZMF:
036D: CD AE 03 CALL     MF      ;Zlecenie *F
0370: 18 CD    JR      MC1
; .COMMENT%
;MD - przegladanie pamieci z mozliwoscia
; modyfikacji.
;*D[POCZATEK][CR]...[CR] - przegl. do przodu
; [SPAC]...[SPAC] - przegl. do tylu
;[LICZ. HEX][CR] LUB [SPAC] - modyfikacja pam.
;ZMIENIA: AF,HL,DE,C STOS: 11
;%
0372:          MD:

```

```

0372: CD F4 01
0375: 40
0376: E7
0377: 43
0378: 7E
0379: DF
037A: 20
037B: CF
037C: 38 16
037E: 2B
037F: 28 F5
0381: 23
0382: FE 10
0384: D0
0385: 4F
0386: D7
0387: 20
0388: CD E1 01
038B: 79
038C: EB
038D: CD F8 01
0390: EB
0391: 73
0392: 30 EA

0394: 23
0395: 18 DF

```

```

0397: 0C
0398: CD 13 02
039B: 40
039C: C1
039D: D1
039E: E1
039F: 71
03A0: CD 3B 02
03A3: 30 FA
03A5: C9

```

```

CALL PARAM ;Pobranie adr. poczat.
DB 40H
SU0: RST LADR ;Wysw. adresu poczat.
DB 43H
LD A,(HL) ;Pobranie komorki pam.
RST LBYTE ;Wyswietlenie (HL)
DB 20H
RST TI1 ;Pobr. pierw. znaku
JR C,SU1 ;Wcisnieto CR
SU2: DEC HL ;Do tylu
JR Z,SU0 ;Wcisnieto SPAC
INC HL ;Odtworzenie HL
CP 10H ;Czy cyfra szesnast.
RET NC ;Inny niz cyfra
LD C,A ;Ochrona cyfry
RST CLR ;Zerowanie wyswietl.
DB 20H
CALL C01 ;Wysw. pirwszej cyfry
LD A,C ;Odtw. cyfry
EX DE,HL ;Ochrona HL
CALL PARA1 ;Pobranie drugiej cyfry
EX DE,HL ;Odtworzenie HL
LD (HL),E ;Ust. nowej wart.
JR NC,SU2 ;Wprow. zakoncz. SPAC
;Wprowadzanie zakonczono klawiszem CR
SU1: INC HL ;Do przodu
JR SU0
;.COMMENT%
;ME - wpisanie stalej do zadanego obszaru pamieci
;*E[OD][SPAC][D0][SPAC][STALA][CR]
;WEJ: C=2
;ZMIENIA: AF,BC,DE,HL STOS: 11
;%
ME: INC C ;3 parametry
CALL EXPR ;Pobranie parametrow
DB 40H
POP BC ;Stala
POP DE ;Do
POP HL ;Od
ME1: LD (HL),C ;Wpisanie stalej
CALL HILO ;Czy DE>=HL
JR NC,ME1 ;DE>=HL
RET
;.COMMENT%
;MF - przegladanie i modyfikacja rej. procesora
;Procedura MF sklada sie z dwoch czesci
;MF: Wyswietla wskazniki sygn. S,Z,H,P,N,C
; ;umozliwiajac latwa modyfikacje Z i CY.
;MF1: Wyswietlanie zawartosci rejestrow A,F,B,C
; ;D,E,H,L,PC,SP,IX,IY z mozliw. modyfikacji.
;
;ZMIENIA: AF,HL,BC,DE STOS: 11
;MF: *F - Wyswietlenie wskaznik. sygnalizacyj.
;Wciskanie klaw. 0-3 powoduje:
;[0] - zerowanie wskaznika Z
;[1] - ustawienie wskaznika Z
;[2] - zerowanie wskaznika CY
;[3] - ustawienie wskaznika CY
;[CR] - wyjście z procedury

```

```

;Wcisniecie nazwy dowolnego rej. spowoduje
;przejscie do MF1:
;MF1: [NAZWA REJ.].... wyswietlanie nazw rej.
;      i ich zawartosci.
;      [CR] - wyjscie z procedury
;[SPAC][NOWA WART.][CR] - ustaw. nowej zawart.
;      wybranego rejestru
;%
03A6: FE 04      CAR:    CP    4      ;Czy ustaw. wskaz. CY
03A8: 30 2F      JR     NC,MF1      ;Nie ustawic
03AA: 1F         RRA          ;CY:=BIT0
03AB: 78         LD     A,B
03AC: 17         RLA          ;Ustawienie CY
03AD: 12      ZAP:    LD     (DE),A      ;Zapamietanie wskaznik.
;Wejscie do procedury MF
;
03AE:           MF:
03AE: D7         RST    CLR      ;Zerow. wyswietlacza
03AF: 70         DB     70H      ;7 mlodszych cyfr
03B0: 21 2E 04   LD     HL,TFLAG ;Tablica wskaznikow
03B3: 11 91 FF   LD     DE,FLOC   ;Adr. rej. F uzytkow.
03B6: 06 08     LD     B,8      ;Licznik przsuniec
03B8: 1A         LD     A,(DE)    ;Pobranie rej. F
03B9: E6 D7     AND     0D7H     ;Maskow. bitow B5,B3
03BB: 17      WYSW:  RLA
03BC: 30 06     JR     NC,ZER     ;Wskaznik=0
03BE: 4E         LD     C,(HL)    ;Pobr. symbolu wskaz.
03BF: F5         PUSH  AF        ;Ochrona AF
03C0: CD AC 01   CALL  COM1      ;Wysw. symbolu
03C3: F1         POP   AF        ;Odtworz. AF
03C4: 23      ZER:  INC     HL      ;Adr. nast. symbolu
03C5: 10 F4     DJNZ  WYSW
03C7: 1A         LD     A,(DE)    ;Pobranie rej. F
03C8: 17         RLA
03C9: 17         RLA          ;Wyizolowanie wsk. Z
03CA: 4F         LD     C,A      ;Dla wskaznika Z
03CB: 1A         LD     A,(DE)    ;Pobr. rej. F uzytkow.
03CC: 1F         RRA          ;Wyizolowanie wsk. CY
03CD: 47         LD     B,A      ;Dla wskaznika CY
03CE: CF         RST    TI1      ;Pobierz znak
03CF: FE 02     CP     2         ;Czy ustaw. wsk. Z ?
03D1: 30 D3     JR     NC,CAR     ;Nie ustawic
03D3: 1F         RRA          ;CY:=bit B0
03D4: 79         LD     A,C
03D5: 1F         RRA
03D6: 1F         RRA          ;Ustawienie wsk. Z
03D7: 18 D4     JR     ZAP      ;Zapamietaj rej. F
;Czesc II  zlecenia *F
03D9:           MF1:
03D9: 57         LD     D,A      ;Zapamiet. nazwy rej.
03DA: D7         RST    CLR      ;Zerow. 7 mlodsz. cyfr
03DB: 70         DB     70H
03DC: 4A         LD     C,D
03DD: CD E0 01   CALL  C0        ;Wysw. nazwy rejestru
03E0: 15         DB     15H
03E1: 7A         LD     A,D
03E2: 21 36 04   LD     HL,ACT1   ;Adr. tab. ACT1
03E5: 01 0C 00   LD     BC,LACT1 ;Dlugosc tablicy
03E8: ED B1     CPIR          ;Przeszukanie ACT1

```



```

03EA: 20 05      JR    NZ,X4 ;Gdy nie znaleziono
03EC: 4E          LD    C,(HL) ;Nazwa rej. z ACT1
03ED: CD AB 01    CALL  COM ;Wyswietl. nazwy
03F0: 15          DB    15H
;Spr. czy ostatnio wcisniety znak jest
;rzeczywiscie nazwa rejestru
03F1: 7A          X4:   LD    A,D ;Przeszuk. tab. ACTBL
03F2: 21 3F 04    LD    HL,ACTBL-3
03F5: 0E 0D        LD    C,NREGS+1 ;Dlugosc ACTBL+1
03F7: 23          X0:   INC    HL
03F8: 23          INC    HL
03F9: 23          INC    HL
03FA: 0D          DEC    C
03FB: C8          RET    Z ;Nazwa falszywa
03FC: BE          CP    (HL)
03FD: 20 F8       JR    NZ,X0
;Nazwa legalna
03FF: CD 1D 04    CALL  DREG ;Wyswiet. zawart. rej.
0402: CF          RST    TI1
0403: D8          RET    C ;Wroc gdy CR
0404: 20 D3       JR    NZ,MF1 ;Inny niz SPAC
;.COMMENT%
;Wcisniwto klawisz SPAC - zmiana zawartosci
;rejestru.%
0406: D7          RST    CLR ;Ust. PWYS=40H
0407: 40          DB    40H
0408: 04          INC    B
0409: 20 02       JR    NZ,BIT16 ;Rej. 16 bitowy
040B: D7          RST    CLR ;Ust. PWYS=20H
040C: 20          DB    20H
040D: CD F5 01    BIT16: CALL  PARAM1 ;Pobranie nowej wart.
0410: D0          RET    NC ;Wroc gdy nie CR
0411: 7D          LD    A,L
0412: 12          LD    (DE),A ;Zapam. mlodszy bajtu
0413: 05          DEC    B
0414: FA 1A 04    JP    M,X8 ;Rej. 8 bitowy
0417: 13          INC    DE ;Rej. 16 bitowy
0418: 7C          LD    A,H ;Starszy bajt
0419: 12          LD    (DE),A ;Zapamietanie
041A: CF          X8:   RST    TI1 ;Pobr. nast. nazwy
041B: 18 BC       JR    MF1
;.COMMENT%
;DREG - procedura pomocnicza zlecenia *F
;Wylicza adres polozenia rej. uzytkownika
;a nastepnie wyswiela jego zawartosc.
;
;WEJ: HL - adres pod ktorym przechowywana jest
; nazwa rej. w tablicy ACTBL
;WYJ: B=0FFH to DE - wskazuje adres rej. 8 bit.
; B=0 to DE - wskazuje adres mniej znacz.
; bajtu rej. 16 bitowego.
;ZMIENIA: AF,HL,DE,B STOS: 9
;%
041D:            DREG:
041D: 16 FF       LD    D,MTOP ;Starszy bajt adr.
041F: 23          INC    HL
0420: 5E          LD    E,(HL) ;Mlodszy bajt adr.
0421: 23          INC    HL
0422: 46          LD    B,(HL) ;B=0 - Rej. 8 bitowy

```

```

                                ;B=1 - Rej.16 bitowy
0423: 1A      LD      A,(DE)      ;Pobr. mlodszy bajtu
0424: DF      RST     LBYTE ;Wysw. mlodszy bajtu
0425: 20      DB      20H
0426: 05      DEC     B
0427: F8      RET     M          ;Gdy rej. 8 bitowy
0428: 13      INC     DE          ;Rej. 16 bitowy
0429: 1A      LD      A,(DE)      ;Pobr. starszy bajtu
042A: DF      RST     LBYTE ;Wysw. starszy bajtu
042B: 22      DB      22H
042C: 1B      DEC     DE          ;Wskazuje mlod. bajt
042D: C9      RET
; .COMMENT%
; TFLAG - tablica wskaznikow sygnalizacyjnych
; zawiera kody siedmiosegmentowe wskaznikow
; sygnal. wysw. na wyswietlaczu.
; %
042E: 6D 5C 00 76  TFLAG: DB      6DH,5CH,0,76H      ;SO-H
0432: 00 73 54 39      DB      0,73H,54H,39H      ; -PNC
; .COMMENT%
; ACT1 - tablica zawiera kody tablicowe (TKLAW)
; oraz kody siedmioseg. rej. S,L,H,IX,IY.
; PRZYKLAD:
; DB      5,6BH ;IX/5
; 5 - kod tablicowy klawisza IX
; 6BH - kod siedmioseg. znaku IX
; %
0436: 05 6B      ACT1:  DB      5,6BH ;IX/5
0438: 06 72      DB      6,72H ;IY/6
043A: 07 6D      DB      7,6DH ;S/7
043C: 08 76      DB      8,76H ;H/8
043E: 09 38      DB      9,38H ;L/9
0440: 10 73      DB      GKLAW,73H ;P/GKLAW
0442:              LACT1 EQU    $-ACT1
; .COMMENT%
; ACTBL - tablica zawierajaca nazwe legalnego
; rejestru (kod tablicowy odpow. klaw.),mniej
; znaczacy bajt adr. wskazujacego polozenie
; zawartosci rej. oraz dlugosc rej.(0-8bitow)
; 1-16bitow)
; %
0442: 0A 92 00      ACTBL: DB      0AH,ALOC&0FFH,0 ;A/0A
0445: 0B 90 00      DB      0BH,BLOC&0FFH,0 ;B/0B
0448: 0C 8F 00      DB      0CH,CLOC&0FFH,0 ;C/0C
044B: 0D 8E 00      DB      0DH,DLOC&0FFH,0 ;D/0D
044E: 0E 8D 00      DB      0EH,ELOC&0FFH,0 ;E/0E
0451: 0F 91 00      DB      0FH,FLOC&0FFH,0 ;F/0F
0454: 08 A6 00      DB      08H,HLOC&0FFH,0 ;H/08
0457: 09 A5 00      DB      09H,LLOC&0FFH,0 ;L/09
045A: 10 A9 01      DB      GKLAW,(PLOC-1)&0FFH,1 ;P/10
045D: 07 97 01      DB      07H,(SLOC-1)&0FFH,1 ;S/07
0460: 05 93 01      DB      5H,(IXLOC-1)&0FFH,1 ;IX/5
0463: 06 95 01      DB      6H,(IYLOC-1)&0FFH,1 ;IY/6
0466:              NREGS EQU    ($-ACTBL)/3
; .COMMENT%
; MG - wejscie do programu uzytkownika
; G[CR] - wejscie w/g aktualnego PC uzytkownika
; G[SPAC][PU1][CR] - wejscie j/w z zastaw. pulapki
; G[SPAC][PU1][SPAC][PU2][CR] - j/w lecz 2 pulapki

```

```

;G[ADRW][CR] - skok do adresu wejścia [ADRW]
;G[ADRW][SPAC][PU1][CR] - j/w z zastaw. pulapki
;G[ADRW][SPAC][PU1][SPAC][PU2][CR]-j/w lecz 2 pul.
;Po napotkaniu ktorejkolwiek z pulapek następuje
;przejście do programu MONITORA i wykonanie
;procedury RESTAR.
;%
0466: F1      MG:      POP    AF      ;Zlikwidowanie adr. powr.
0467: CD 07 00      CALL   TI      ;Pobr. pierwszego znaku
046A: 40          DB      40H
046B: 28 06          JR      Z,GOA  ;CR lub SPAC
046D: CD F8 01      CALL   PARA1  ;Pobranie ADRW
0470: 22 A9 FF      LD      (PLOC-1),HL ;PC := ADRW
0473: 38 34      GOA:    JR      C,GO4 ;Wcisnieto CR
;Pobranie 1 lub 2 pulapek
0475: 0E 40      GO1:    LD      C,KRESKA ;Symbol pulapki
0477: CD AB 01      CALL   COM     ;Wyswietlenie symbolu
047A: 14          DB      14H
047B: 06 02          LD      B,2    ;Max. 2 pulapki
047D: CD F4 01      PU2:    CALL   PARAM ;Pobranie pulapki
0480: 40          DB      40H
0481: E5          PUSH   HL      ;Zapamietanie pulapki
0482: 05          DEC     B       ;Nie zmienia CY !
0483: 38 11          JR      C,TRA1 ;CR -zastaw pobr. pulap.
0485: 20 F6          JR      NZ,PU2
;Obsluga błedu systemowego
0487:      ERROR:
0487: 31 8D FF      LD      SP,TOS   ;Stos systemowy
048A: D7          RST     CLR     ;Zerowanie wyswietl.
048B: 80          DB      80H
048C: 21 34 00      LD      HL,K02   ;Adr. komunikatu "Err"
048F: CD D4 01      CALL   PRINT ;Wyswiet. "Err"
0492: 35          DB      35H
0493: C3 75 02      JP      START1 ;Pobierz kolejne zlec.
;.COMMENT%
;Zastawienie 1 lub 2 pulapek
;Zastawienie pulapki polega na sztucznym wsta-
;wieniu do programu uzytkownika rozkazu RST 30H.
;Wykonanie tego rozkazu przez program uzytko-
;wnika spowoduje przejście do procedury RESTAR
;programu MONITORA. Procedura ta odtwarza komorki
;pamieci do ktorych wpisano w sposob sztuczny
;rozkaz RST 30H. Aby odtworzenie bylo zadaniem
;wykonywalnym zarowno adres pulapki jak i zawa-
;rtosc komorki pamieci wskazywana przez ten
;adres musza zostac zapamietane - co realizuje
;ponizszy ciag rozkazow. Pulapke mozna ustawic
;wylacznie w pamieci typu RAM.
;Informacja niesiona przez rej. B:
;B=0 - bylo dwie pulapki
;B=1 - byla jedna pulapka
;%
0496: 21 AB FF      TRA1:   LD      HL,TLOC   ;Adr. przechowyw. pulap.
0499:      TRA2:
0499: D1          POP     DE      ;Adres pulapki
049A: 73          LD      (HL),E
049B: 23          INC     HL
049C: 72          LD      (HL),D   ;Zapamietanie adr. pulap.
049D: 23          INC     HL

```

049E: 1A	LD A,(DE)
049F: 77	LD (HL),A ;Zap. komorki pamieci
04A0: 3E F7	LD A,RST30 ;Rozkaz RST 30H
04A2: 12	LD (DE),A ;Zastawienie pulapki
04A3: 23	INC HL
04A4: 78	LD A,B
04A5: 04	INC B
04A6: B7	OR A ;Ustawienie wskaznikow
04A7: 28 F0	JR Z,TRA2 ;Ustaw 2 pulapke
04A9: D7	G04: RST CLR ;Wygaszenie wyswietlacza
04AA: 80	DB 80H
	;GSTAT=0 - sygnalizuje wykonywanie progr. uzytkow
04AB: AF	G05: XOR A ;Zerowanie A
04AC: 32 B3 FF	LD (GSTAT),A ;Zaznacz. prog. uzytkow.
04AF: D3 FC	OUT (RESI),A ;Kasow. zglosz. przerwan.
04B1: C3 99 FF	JP EXIT ;Wejscie do prog. uzytkow.
	;.COMMENT%
	;M3 - wymiana rejestrow procesora
	;Wymiana rejestrow glownych na pomocnicze i odwr.
	;ZMIENIA: AF,HL,DE,BC STOS: 9
	;*3[CR]
	;%
04B4: CF	M3: RST TI1 ;Czy CR
04B5: 30 D0	JR NC,ERROR ;Nie CR
	;Pobranie rej. glownych uzytkownika
04B7: 31 8D FF	LD SP,ELOC
04BA: D1	POP DE ;Pobranie DE
04BB: C1	POP BC ;BC
04BC: F1	POP AF ;Pobranie AF
04BD: 2A A5 FF	LD HL,(LLOC) ;Pobranie HL
	;Wymiana na rej. pomocnicze
04C0: 08	EX AF,AF'
04C1: D9	EXX
	;Odtworzenie rejestrow glownych uzytkownika
04C2: 22 A5 FF	LD (LLOC),HL ;Odtworzenie HL
04C5: F5	PUSH AF
04C6: C5	PUSH BC
04C7: D5	PUSH DE ;Odtworzenie AF,BC,DE
04C8: C3 70 02	JP START ;Ustaw. SP w START.
	;.COMMENT%
	;M7 - inicjacja systemu lub ustaw. parametrow
	; transmisji magnetofonowej.
	;A. Inicjacja systemu CA80 (skok do adr. 0000H)
	;*7[CR]
	;B. Ustawienie parametrow transmisji magnetof.
	;*7[MAGSP DLUG][CR]
	;DLUG - dlugosc bloku danych <1...0FFH>
	;MAGSP - szybkosc transmisji magnetofonowej.
	;%
04CB: D7	M7: RST CLR ;Ustawienie PWYS
04CC: 40	DB 40H
04CD: CF	RST TI1
04CE: DA 00 00	JP C,CA80 ;Inicjacja systemu
04D1: FE 10	CP 10H ;Tylko cyfry szesnastkow.
04D3: 30 B2	JR NC,ERROR
04D5: CD F8 01	CALL PARA1 ;Pobranie parametrow
04D8: 30 AD	JR NC,ERROR ;Gdy SPAC
04DA: 22 B1 FF	LD (DLUG),HL ;Zapisanie paramet
04DD: C9	RET

```

; .COMMENT%
; M9 - poszukiwanie slowa 16-to bitowego
;      lub 8-mio bitowego.
; *9[SLOW0][SPAC][ADRES POCZATKU][CR]
; Jesli bardziej znaczacy bajt [SLOW0]#0 to
; poszukiwanie slowa 16-to bitowego.
; Jesli bardziej znaczacy bajt [SLOW0]=0 to
; poszukiwanie slowa 8-mio bitowego.
;
; Zlecenie poszukuje [SLOW0] w 16kb pamieci
; poczawszy od adresu poczatkowego. Po znalezie-
; niu wywolana zostaje procedura *D. Powrot do
; procedury *9 i poszukiwanie dalszych slow
; po naciśnięciu klawisza "G".
; %
04DE: CD 13 02      M9:      CALL  EXPR  ;Pobierz dwa parametry
04E1: 40            DB      40H
04E2: 01 00 40      LD      BC,4000H ;Obszar 16kb
04E5: E1            POP     HL      ;Adres poczatku
04E6: D1            M91:     POP     DE    ;Slowo do znalezienia
04E7: 7A            M90:     LD      A,D
04E8: B7            OR      A      ;Czy slowo 16 bitow ?
04E9: 20 01         JR      NZ,SLOW16 ;Slowo 16 bitowe
04EB: 7B            LD      A,E     ;Slowo 8 bitowe
04EC: ED B1         SLOW16: CPIR      ;Poszukiw. pierwsz. bajtu
04EE: E0            RET     PO      ;Nie znaleziono
04EF: 7A            LD      A,D
04F0: B7            OR      A      ;Czy slowo 8 bitowe
04F1: 28 04         JR      Z,SLOW8  ;Slowo 8 bitowe
04F3: 7B            LD      A,E     ;Slowo 16 bitowe
04F4: BE            CP      (HL)    ;Spraw. 2 bajtu
04F5: 20 F0         JR      NZ,M90   ;Drugi bajt do kitu
04F7: 2B            SLOW8: DEC     HL  ;Na pierwszy bajt
04F8: D5            PUSH    DE      ;Ochrona DE
04F9: CD 76 03      CALL    SU0     ;Wywołanie zlec. *D
04FC: 23            INC     HL      ;Szuk. dalsze slowa
04FD: 18 E7         JR      M91
; .COMMENT%
; MB - przesuniecie obszaru pamieci
; *B[ADR1][SPAC][ADR2][SPAC][ADR3][CR]
; Zlecenie powoduje przesuniecie obszaru
<ADR1,ADR2>
; do obszaru rozpoczynajacego sie od adresu ADR3.
; Przesuwanie jest inteligentne tzn. ADR3 moze
; lezec zarowno wewnatrz <ADR1,ADR2> jak i poza
; tym obszarem.(przesow zawsze poprawny)
; Musi byc spelnione: ADR1=<ADR2 - w przeciwnym
; razie zlecenie sygnalizuje blad.
; %
04FF: 0C            MB:      INC     C      ;3 parametry
0500: CD 13 02      CALL    EXPR  ;Pobranie parametrow
0503: 40            DB      40H
0504: C1            POP     BC      ;ADR3
0505: E1            POP     HL      ;ADR2
0506: D1            POP     DE      ;ADR1
0507: B7            OR      A      ;CY=0
0508: E5            PUSH    HL      ;ADR2
0509: ED 52         SBC     HL,DE   ;ADR2-ADR1
050B: DA 87 04      JP      C,ERROR ;Gdy ADR2<ADR1

```

```

050E: E3      EX      (SP),HL      ;(SP) - dlugosc
050F: E5      PUSH   HL      ;ADR2
0510: D5      PUSH   DE      ;ADR1
0511: ED 42    SBC     HL,BC ;ADR2-ADR3
0513: 38 11    JR      C,PRZOD      ;ADR2<ADR3
0515: E1      POP     HL
0516: E5      PUSH   HL      ;ADR1
0517: ED 42    SBC     HL,BC ;ADR1-ADR3
0519: 30 0B    JR      NC,PRZOD ;ADR1>=ADR3
;ADR1< ADR3< ADR2
;Przesuwanie do tylu
051B: E1      POP     HL      ;ADR1
051C: D1      POP     DE      ;ADR2
051D: E1      POP     HL
051E: E5      PUSH   HL      ;Dlugosc
051F: 09      ADD     HL,BC ;Dlugosc+ADR3
0520: EB      EX      DE,HL ;DE - Dlugosc+ADR3
;HL - ADR2
0521: C1      POP     BC      ;Dlugosc
0522: 03      INC     BC      ;Dlugosc+1
0523: ED B8    LDDR
0525: C9      RET
0526:
PRZOD:
0526: E1      POP     HL      ;ADR1
0527: 59      LD      E,C
0528: 50      LD      D,B ;DE := ADR3
0529: C1      POP     BC      ;ADR2
052A: C1      POP     BC      ;Dlugosc
052B: 03      INC     BC      ;Dlugosc+1
052C: ED B0    LDIR
052E: C9      RET
;.COMMENT%
;MWCIS - procedura bezwarunkowego przejścia do
;poczatku petli glownej programu
;MONITORA (etykieta START).
;Przejscie z programu uzytkownika do programu
;MONITORA mozna wymusic w dowolnej chwili
;wciskajac klawisz "M". Jesli procedurra obsługi
;przerwania NMI stwierdzi, ze klawisz "M" jest
;wcisniety to nastepuje skok do przedstawionej
;nizej procedury MWCIS.
;%
052F:
MWCIS:
052F: F3      DI          ;Blokada przerwan
;Inicjacja obszaru RAM <APWYS,NMIU>
0530: 21 FD 05  LD      HL,TNMIU
0533: 11 CC FF  LD      DE,NMIU
0536: 01 0C 00  LD      BC,LIOCA
0539: ED B8    LDDR
;(GSTAT)=0 - wykonywany program uzytkownika
;(GSTAT)#0 - wykonywany program MONITORA
053B: 3A B3 FF  LD      A,(GSTAT)
053E: B7      OR      A
053F: C2 70 02 JP      NZ,START ;Wyk. program MONITORA
;Wykonywany program uzytkownika
;Odtw. rej. angazowanych w procedurze NMI
0542: C1      POP     BC
0543: D1      POP     DE
0544: E1      POP     HL

```

```

0545: F1          POP    AF      ;Odtw. AF,HL,DE,BC
; .COMMENT%
;RESTAR - procedura przejścia z programu uzytkow.
;do programu MONITORA.
;Procedura powoduje:
;1. Zapamiętanie stanu procesora uzytkownika
;w obszarze pamieci <TOS,PLOC>
;2. Likwidację wszystkich (1 lub 2) pulapek.
;Jesli ktorakolwiek z pulapek byla zastawiona
;to następuje dobicie kropki do najstarszej
;cyfry wyswietlacza i oczekiwanie na wcisniecie
;dowolnego klawisza.
;Wcisniecie spowoduje skok do etykiety "START"
;Jesli zadna z pulapek nie byla zastawiona to
;następuje badanie stanu komorki pamieci
;o adresie "KROK/FFA2"
;3. Jesli (KROK)=0 to skok do etykiety START
;4. Jesli (KROK)#0 to (KROK):=0 i (KROK+1):=0
; - likwidacja pracy krokowej po czym skok
;do etykiety MC1: (zlecenie MC)
;%
;Schowanie stanu procesora uzytkow. na stos
0546:          RESTAR:
0546: E5          PUSH   HL
0547: D5          PUSH   DE
0548: C5          PUSH   BC
0549: F5          PUSH   AF
054A: DD E5       PUSH   IX
054C: FD E5       PUSH   IY      ;Schow. IY,IX,AF..HL
; <EXIT-1,TOS> - obszar przechowyw. rej. uzytkow.
054E: 11 99 FF   LD      DE,EXIT      ;Adr. poczatkowy
0551: 7A          LD      A,D      ;A#0
; (GSTAT)#0 - wykonyw. jest program MONITORA
0552: 32 B3 FF   LD      (GSTAT),A
; .COMMENT%
;SP - wsk. mlodszy bajt IY uzytkow. (patrz wyzej)
;SP+11 - wskazuje rej. H
;SP+12 - wsk. mlodszy bajt rej. PC uzytkownika
;SP+13 - wsk. starszy bajt rej. PC uzytkownika
;SP+14 - stos uzytkownika przed napotkaniem
; pulapki lub rozkazu RST 30H !
;%
0555: 21 0E 00   LD      HL,14
0558: 39          ADD     HL,SP ;HL- stos uzytkownika !
0559: EB          EX      DE,HL ;HL=EXIT
;DE=stos uzytkownika
;Zapamiętanie SP,IY,IX,AF,BC,DE uzytkownika
;W obszarze <EXIT-1,ELOC>
055A: 06 06       LD      B,6      ;Gdyz SP,IY,IX,AF,BC,DE
055C: 2B          RST0:  DEC     HL
055D: 72          LD      (HL),D      ;Starszy bajt
055E: 2B          DEC     HL
055F: 73          LD      (HL),E      ;Mlodszy bajt
0560: D1          POP     DE      ;Kolej. IY,IX,AF,BC,DE,HL
0561: 10 F9       DJNZ    RST0
;DE - zawartosc rej. HL uzytkownika
;SP - wskazuje PC uzytkownika !
;HL - wskazuje komorke pamieci o adr. ELOC=TOS !
0563: C1          POP     BC      ;Rejestr PC uzytkownika

```

```

0564: F9          LD    SP,HL ;Ustawienie stosu system.
                   ;HL=TOS
0565: 2E A5        LD    L,LLOC&0FFH ;HL=adr. LLOC
                   ;Zapamietanie rej. HL uzytkownika
0567: 73          LD    (HL),E      ;Mlodszy bajt HL
0568: 23          INC    HL
0569: 72          LD    (HL),D      ;Starszy bajt HL
                   ;.COMMENT%
                   ;BC - zawiera PC uzytkownika
                   ;Jesli napotkano pulapke to rej. PC nalezy zmniejsz.
                   ;o 1, gdyz PC wskazuje wowczas pierwsza komorke
                   ;pamieci po rozkazie RST 30H a powinien wskazywac
                   ;rozkaz RST 30H !
                   ;%
056A: 0B          DEC    BC      ;Zalozenie ze pulapka
056B: 2E AB        LD    L,TLOC&255 ;HL=TLOC
                   ;.COMMENT%
                   ;Kryterium odkrycia pulapki jest nastepujace:
                   ;Jesli przyczyna wejscia do RESTAR byla pulapka
                   ;to musi byc spelnione:
                   ;BC=(TLOC+1)(TLOC) - PULAPKA 1
                   ;lub
                   ;BC=(TLOC+4)(TLOC+3) - PULAPKA 2
                   ;Sprawdzenie powyzzszego warunku.
                   ;%
056D: 16 02        LD    D,2      ;Sprawdz. 2 pulapek
056F: 7E          POWTR: LD    A,(HL)
0570: A9          XOR    C      ;Mlodsze bajty rowne ?
0571: 23          INC    HL
0572: 20 04        JR    NZ,NIER    ;Nie rowne
0574: 7E          LD    A,(HL)
0575: A8          XOR    B      ;Starsze bajty rowne ?
0576: 28 06        JR    Z,RST1     ;Przycz. byla pulapka !
0578: 23          NIER: INC    HL
0579: 23          INC    HL      ;HL=TLOC+3
057A: 15          DEC    D
057B: 20 F2        JR    NZ,POWTR
                   ;.COMMENT%
                   ;Przyczyna wejscia do RESTAR bylo wcisniecie
                   ;klaw. "M" lub wykonanie rozkazu RST 30H usta-
                   ;wionego przez uzytkownika - nie przez zlec.*G !
                   ;%
057D: 03          INC    BC      ;Odtworzenie PC uzytk.
                   ;Schowanie PC uzytkownika
057E:          RST1:
057E: 2E A9        LD    L,(PLOC-1)&255
0580: 71          LD    (HL),C      ;Mlodszy bajt PC
0581: 23          INC    HL
0582: 70          LD    (HL),B      ;Starszy bajt PC
                   ;.COMMENT%
                   ;Informacja niesiona przez rej. D jest nastep:
                   ;D=0 - nie bylo pulapki
                   ;D=1 - byla pulapka nr. 2
                   ;D=2 - byla pulapka nr.1
                   ;%
                   ;Kasowanie pulapek
0583: 1E 02        LD    E,2      ;2 pulapki
0585: 23          INC    HL      ;Wskazuje TLOC
0586: 4E          TRP:  LD    C,(HL)

```



```

0587: AF      XOR    A      ;Zerowanie A
0588: 77      LD     (HL),A    ;Zerowanie
0589: 23      INC     HL
058A: 46      LD     B,(HL)      ;BC-adr. pulapki
058B: 77      LD     (HL),A      ;Zerowanie
058C: 23      INC     HL
058D: 7E      LD     A,(HL)
058E: 02      LD     (BC),A      ;Odtw. (PULAPKA)
058F: 23      INC     HL      ;Wskazuje TLOC+3
0590: 1D      DEC     E
0591: 20 F3    JR      NZ,TRP
; .COMMENT%
; Jesli pulapek nie bylo to powyjsza petla spo-
; woduje niszczodliwe, dwukrotne wpisanie informa-
; cji do komorki pam. o adr. 0000H. W CA80 jest
; to obszar pamieci stalej - EPROM.
;%
0593: 7A      LD     A,D
0594: CD BD 05 CALL    EMUL ;Spraw. czy emulator
0597: 7A      LD     A,D
0598: B7      OR     A
0599: 20 17    JR      NZ,PUL      ;Byla pulapka
059B: 3A A2 FF LD     A,(KROK)
059E: B7      OR     A
059F: CA 70 02 JP     Z,START      ;Nie praca krokowa
; Praca krokowa
; .COMMENT%
; ZCHAN/3H - slowo sterujace dla kanalu 0 ukladu
; Z80A CTC. Powoduje zerowanie "TIMERA".(MIK04)
;%
05A2: 3E 03    LD     A,ZCHAN
05A4: D3 F8    OUT    (CHAN0),A ;Zerow. kanalu 0
; Likwidacja pracy krokowej
05A6: 21 00 00 LD     HL,0
05A9: 22 A2 FF LD     (KROK),HL ;Likwidacja
; Realizacja skoku do MC1 (zlec. MC) z jednocz.
; wykonaniem rozkazu RETI.
05AC: 21 3F 03 LD     HL,MC1
05AF: E5      PUSH   HL      ;Na stos
05B0: ED 4D    RETI      ;Dla potrzeb Z80A CTC !
; Czekanie na wcisniecie klawisza -byla pulapka
05B2: PUL:
05B2: 21 FE FF LD     HL,BWYS+7 ;Najstarsza cyfra
05B5: CB FE    SET    KROP,(HL) ;Zaswiecenie kropki
05B7: CD 84 01 CALL   CIM ;Czek. na wcis. klaw.
05BA: C3 70 02 JP     START ;Do wejsc. glownego
05BD: .LIST
; .COMMENT%      Zmiany w EMUL - MIK08 str.1-34
; EMUL - skok do emulatora pod warunkiem,ze PA0=1.
; Wywołanie EMUL następuje z procedury RESTAR
; lub START.
; REJ. A=<0,3> - wywołanie z procedury RESTAR
; A=0 - nie bylo pulapki
; A=1 - byla pulapka nr.2
; A=2 - byla pulapka nr.1
; Rej. A=0FFH - wywołanie z procedury START
;%
05BD: EMUL:
05BD: 32 E7 FF LD     (LCI-1),A ;Zapamietanie rej A

```

```

05C0: DB F0          IN    A,(PA)          ;Port 8255
05C2: E6 01          AND    1              ;Czy PA0=1 ?
05C4: C8              RET    Z              ;Gdy PA0=0
05C5: C3 BB FF        JP     EM             ;Skok do emulatora
; .XLIST
;.COMMENT%
;Tablica inicjacji obszaru RAM angazowanego
;przez CA80.
;%
05C8:
TRAM:
05C8: 66 FF          DW     TOS-27H        ;Stos uzytkownika
;Wejscie do programu uzytkownika
05CA: D1             POP    DE             ;EXIT
05CB: C1             POP    BC
05CC: F1             POP    AF
05CD: DD E1          POP    IX
05CF: FD E1          POP    IY
05D1: E1             POP    HL
05D2: F9             LD     SP,HL
05D3: 00             NOP
;KROK
05D4: 00             NOP
05D5: 21 00 C1        LD     HL,HLUZYT
05D8: FB             EI
05D9: C3 00 C0        JP     PCUZYT
;Pulapki
05DC: 00 00          DW     0             ;Pulapka1
05DE: 00             DB     0
05DF: 00 00          DW     0             ;Pulapka2
05E1: 00             DB     0
;Parametry wspolpracy z magn.
05E2: 10             DB     16            ;DLUG
05E3: 25             DB     25H           ;MAGSP
;Klucze programowe
05E4: FF             DB     -1            ;GSTAT
05E5: FF             DB     -1            ;ZESTAT
;Skoki posrednie
05E6: C3 00 08        JP     800H        ;M8
05E9: C3 87 04        JP     ERROR    ;ERRMAG
05EC: C3 06 08        JP     806H        ;EM
05EF: C3 03 08        JP     803H        ;RTS
;Systemowe skoki posrednie
;Inicjowane wraz z wcisnieciem klaw. "M"
IOCA: DW    PWYS      ;APWYS
      JP    CSTSM     ;CSTS
      JP    CIM        ;CI
      JP    RESTAR     ;AREST
TNMIU: RET
LIOCA: EQU    $-IOCA
      DW    0          ;c.d. NMIU
      JP    ERROR     ;INTU
KTRAM: EQU    $-1
LTRAM: EQU    $-TRAM
;
0603:
      .LIST
;.COMMENT%
;Wejscie do EPROM uzytkownika U11
;po nacisnieciu *89 w CA80
;Adres wejscia do WEJU11/603
;pod adresem 0836H - MIK06 str.139

```

```

0603: 3A 01 80
0606: FE AA
0608: C2 87 04
060B: C3 02 80
060E: FF FF FF FF
0612: FF FF FF FF
0616: FF FF FF FF
061A: FF FF FF

; %
WEJU11: LD    A,(8001H) ;Bajt testujacy
        CP    0AAH ;Czy jest EPROM ?
        JP    NZ,ERROR ;Brak EPROM
        JP    8002H ;Jest EPROM w U11
        DW    -1,-1
        DW    -1,-1
        DW    -1,-1
        DB    -1,-1,-1
;
; .XLIST
; *****
; O B S L U G A M A G N E T O F O N U
; *****
; .COMMENT%
;LSYNCH - ilosc bajtow synchronizacji
;MARK - wyroznik poczatku rekordu
;ILPR - ilosc probek dla pol bitu
; %
061D: LSYNCH EQU 20H
061D: MARK EQU 0E2FDH
061D: ILPR EQU 20
061D: LOW1 EQU ILPR-ILPR/2-1
061D: HIG1 EQU ILPR+ILPR/2-1
061D: LOW2 EQU 2*ILPR-ILPR/2-1
061D: HIG2 EQU 2*ILPR+ILPR/2-1
;
; .COMMENT%
;M4 - zapis na magnetofon
;Zlecenie powoduje zapisanie na magnetofon obsza-
;ru pamieci <ADR1,ADR2> i nadanie mu nazwy [NAZWA]
;*4[ADR1][SPAC][ADR2][SPAC][NAZWA][CR]
; %
061D: 0C
061E: CD 13 02
0621: 40
0622: C1
0623: 41
0624: D1
0625: E1
M4: INC C ;3 Parametry
CALL EXPR
DB 40H
POP BC
LD B,C ;B - nazwa
POP DE ;ADR2
POP HL ;ADR1
; .COMMENT%
;ZMAG - procedura systemowa
; Zapisanie obszaru pamieci na magnetofon.
;WEJ: B - nazwa
; HL - ADR1
; DE - ADR2
;WYJ: obszar <ADR1,ADR2> zapisany zostanie na
;magnetofon pod nazwa [NAZWA].
;ZMIENIA: AF,HL,C STOS: 13
;
;Budowa rekordu:
;MARK,NAZWA,DLUG,ADRES,-SUMN,BLOK DANYCH,-SUMD.
;MARK - wyroznik poczatku REKORDU
;NAZWA - nazwa zbioru zapisywanego na magnet.
;DLUG - dlugosc bloku danych
;SUMN - suma kontrolna naglowka <NAZWA,ADRES>
;SUMD - suma kontrolna bloku danych.
;<PIERWSZY BAJT,OSTATNI BAJT BLOKU DANYCH>
; %

```

0626:		ZMAG:	
0626: CD 97 06		CALL	SYNCH ;Bajty synchronizacji
0629: C5		PUSH	BC ;Nazwa na stosie
062A: E5		WR0:	PUSH HL ;Ochrona HL
062B: 3A B1 FF		LD	A,(DLUG) ;Dlugosc bloku danych
062E: 4F		LD	C,A
062F: 06 00		LD	B,0
			;Wyliczanie dlugosci bloku danych. Ostatni blok
			;moze byc krotszy niz "DLUG".
0631: 04		WR1:	INC B
0632: 0D		DEC	C
0633: 28 05		JR	Z,WR2
0635: CD 3B 02		CALL	HILO ;HL:=HL+1 i DE-HL
0638: 30 F7		JR	NC,WR1 ;DE>=HL
			;Rej. B - wyliczona dlugosc bloku danych.
063A: D5		WR2:	PUSH DE ;Ochrona DE
063B: 21 FD E2		LD	HL,MARK ;Wyroznik rekordu
063E: CD A2 06		CALL	PADR ;Zapisanie MARK na mag.
0641: D1		POP	DE ;ADR2
0642: E1		POP	HL ;ADR1
0643: F1		POP	AF
0644: F5		PUSH	AF ;A - nazwa
0645: D5		PUSH	DE ;Ochrona DE
0646: 5F		LD	E,A ;Nazwa
0647: 16 00		LD	D,0 ;Zerow. sumy kontrolnej
0649: CD A7 06		CALL	PBYT ;Zapisanie nazwy
064C: 7B		LD	A,E ;Nazwa
064D: DF		RST	LBYTE ;Wyswietlenie nazwy
064E: 25		DB	25H
064F: 78		LD	A,B ;Dlug. bloku danych
0650: CD A7 06		CALL	PBYT ;DLUG - na magnetofon
0653: CD A2 06		CALL	PADR ;Zapis. adresu ladowania
0656: E7		RST	LADR ;Wysw. adr. ladowania
0657: 40		DB	40H
0658: AF		XOR	A ;Zerowanie A
0659: 92		SUB	D ;-SMUN - sum kontr. nagl.
065A: CD A7 06		CALL	PBYT ;Zap. -SUMN
065D: 16 00		LD	D,0 ;Zerowanie SUMD
			;Zapisanie bloku danych na magnetofon
065F: 7E		WR3:	LD A,(HL) ;Pobranie danej
0660: CD A7 06		CALL	PBYT ;Zapisanie danej
0663: 23		INC	HL ;Adr. nastepnej danej
0664: 10 F9		DJNZ	WR3
			;SUMD - suma kontrolna bloku danych.
0666: AF		XOR	A ;A=0
0667: 92		SUB	D ;-SUMD
0668: CD A7 06		CALL	PBYT ;-SUMD na magnetofon
066B: D1		POP	DE ;Adr. konca
066C: 2B		DEC	HL
066D: CD 3B 02		CALL	HILO ;Czy DE>=HL
0670: 30 B8		JR	NC,WR0 ;DE>=HL
			;DE<HL - koniec zapisywania
0672: C1		POP	BC ;Zdjecie nazwy
0673: C9		RET	
			;.COMMENT%
			;M5 - zapisanie rekordu EOF na magnetofon
			;*5[ADR.WEJ.][SPAC][NAZWA][CR]
			;ADR.WEJ. - adres wejscia do programu o nazwie
			;[NAZWA]

```

;Budowa rekordu EOF:
;MARK,NAZWA,DLUG=0,ADR.WEJ.,-SUMN
;ZMIENIA: AF,D,C          STOS: 11
;%
0674: CD 13 02      M5:      CALL  EXPR  ;Pobranie 2 parameteow
0677: 40             DB      40H
0678: C1             POP     BC
0679: 41             LD      B,C   ;Nazwa
067A: E1             POP     HL   ;Adr. wejscia
; .COMMENT%
;ZEOF - procedura systemowa
;Zapisanie rekordu EOF na magnetofon.
;WEJ: HL - adres wejscia do programu o nazwie
;      podanej w rej. B.
;      B - nazwa programu
;WYJ: zapisanie rekordu EOF
;ZMIENIA: AF,C,D          STOS: 7
;%
067B: E5      ZEOF:      PUSH  HL   ;Ochrona HL
067C: CD 97 06      CALL  SYNCH ;Bajty synchronizacji
067F: 21 FD E2      LD      HL,MARK ;Wyroznik pocz. rekordu
0682: CD A2 06      CALL  PADR  ;MARK - na magnet.
0685: 78             LD      A,B   ;Nazwa
0686: 16 00             LD      D,0   ;Zerow. sumy kontrolnej
0688: CD A7 06      CALL  PBYT  ;Zapis. nazwy
068B: AF             XOR     A     ;A=0
068C: CD A7 06      CALL  PBYT  ;DLUG=0
068F: E1             POP     HL   ;Odtw. ADR. WEJ.
0690: CD A2 06      CALL  PADR  ;ADR. WEJ. - na magnet.
0693: AF             XOR     A     ;A=0
0694: 92             SUB     D     ; -SUMN
0695: 18 10             JR      PBYT ; -SUMN - na magnet.
; .COMMENT%
;SYNCH - procedura pomocnicza
;Zapisuje rekord synchronizacji bedacy ciagiem
;32 bajtow o wartosci 00H.
;WEJ: -
;WYJ: zapis 32 bajtow 00H na magnetofon.
;ZMIENIA: AF          STOS: 6
;%
0697: C5      SYNCH:      PUSH  BC   ;Ochrona BC
0698: 06 20      LD      B,LSYNCH ;Ilosc bajtow
069A: AF      PBX:      XOR     A     ;A=0
069B: CD AB 06      CALL  PBYTE
069E: 10 FA      DJNZ  PBX
06A0: C1      POP     BC
06A1: C9      RET
; .COMMENT%
;PADR - procedura pomocnicza
; Zapis rej. HL na magnetofon.
;WEJ: HL - dana do zapisania
;WYJ: zapisanie stanu rej. HL na magnetofon
; D - aktualny stan sumy kontrolnej.
;ZMIENIA: AF,C,D          STOS: 6
;%
06A2: 7D      PADR:      LD      A,L
06A3: CD A7 06      CALL  PBYT  ;Zapisanie rej. L
06A6: 7C      LD      A,H
;Zapisanie rej. H

```

```

; .COMMENT%
; PBYT - procedura pomocnicza
; Zapisanie rej. A na magnetofon.
; Obliczanie sumy kontrolnej w rej. D [D:=D+A]
; WEJ: A - dana do zapisania
; WYJ: zapisanie rej. A na magnet.
; D - aktualny stan sumy kontrolnej
; ZMIENIA: AF,C,D          STOS: 5
;%
06A7: 4F      PBYT: LD C,A ;Ochrona rej. A
06A8: 82      ADD A,D ;Suma modulo256
06A9: 57      LD D,A ;Suma kontrolna
06AA: 79      LD A,C ;Odtw. rej. A
; .COMMENT%
; PBYTE - procedura pomocnicza
; Działanie jak PBYT lecz nie jest obliczana
; suma kontrolna.
; ZMIENIA: AF          STOS: 5
;%
06AB: D5      PBYTE: PUSH DE
06AC: C5      PUSH BC ;Ochrona DE,BC
06AD: 4F      LD C,A ;Dana do zapisania
06AE: 1E 09   LD E,9 ;Ilosc bitow
06B0: CD E7 06 BIT1: CALL GJED ;Generowanie jedyнки
06B3: CD DC 06 BIT4: CALL GZER ;Generowanie zera
06B6: 1D      BIT3: DEC E
06B7: 28 18   JR Z,KBIT ;Koniec zapisu
06B9: 79      LD A,C
06BA: 1F      RRA ;CY:=bit0
06BB: 4F      LD C,A
06BC: 38 F2   JR C,BIT1 ;Gdy jedyнка
06BE: CD DC 06 CALL GZER ;Generowanie zera
06C1: 79      LD A,C
06C2: 1F      RRA
06C3: 38 05   JR C,BIT2
06C5: CD E7 06 CALL GJED ;Generowanie jedyнки
06C8: 18 EC   JR BIT3
06CA: 4F      BIT2: LD C,A
06CB: CD FE 06 CALL GJEDD ;Gener. podwojnej jedyнки
06CE: 1D      DEC E ;Na pewno jest E#0
06CF: 18 E2   JR BIT4
; Generowanie 2 bitow stopu
06D1: 16 04   KBIT: LD D,4
06D3: CD DC 06 KBIT1: CALL GZER ;Generowanie zera
06D6: 15      DEC D
06D7: 20 FA   JR NZ,KBIT1
06D9: C1      POP BC
06DA: D1      POP DE ;Odtw. BC,DE
06DB: C9      RET
; .COMMENT%
; GZER - generowanie zera
; Na wyjściu magnetofonowym wymuszony zostanie
; stan 0 trwajacy 20 probek.
; ZMIENIA: AF,B          STOS: 1
;%
06DC: 06 14   GZER: LD B,ILPR ;Ilosc probek
06DE: CD 09 07 CALL RESMAG ;Zerowanie wyjścia
06E1: CD 02 07 GZE1: CALL DEL02 ;Opóźnienie
06E4: 10 FB   DJNZ GZE1

```

06E6: C9

RET

```
;.COMMENT%
;GJED - generowanie jedynki
;Na wyjściu magnet. wymuszony zostaje stan 1
;trwający 16 próbek i stan 0 trwający 4 próbki.
;Razem: 20 próbek
;Powyższy kształt jedynki ustalono w sposób dos-
;wiadczalny dążąc do uzyskania jak najlepszych
;efektów.
;ZMIENIA: AF,B          STOS: 2
;%
```

06E7: 06 10
06E9: 3E 10
06EB: 32 F4 FF
06EE: D3 F0
06F0: 3E 09
06F2: D3 F3
06F4: CD E1 06
06F7: CD 09 07
06FA: 06 04
06FC: 18 E3

```
GJED:  LD  B,ILPR-4
GJED1: LD  A,10H ;Bit B4 - magnet.(MIK94)
      LD  (KLA),A
      OUT (PA),A ;Dla płytki MIK94
      LD  A,9
      OUT (CONTR),A ;Dla płytki MIK90
      CALL GZE1
      CALL RESMAG ;Zerow. wyjścia magnet.
      LD  B,4
      JR  GZE1 ;4 próbki = 0
```

```
;.COMMENT%
;GJEDD - generowanie podwójnej jedynki
;na wyjściu magnet. wymuszony zostanie stan 1
;trwający 2*ILPR-4=36 próbek i stan 0 trwający
;4 próbki.
;ZMIENIA: AF,B          STOS: 2
;%
```

06FE: 06 24
0700: 18 E7

```
GJEDD: LD  B,2*ILPR-4
      JR  GJED1
```

```
;.COMMENT%
;DEL02 - realizacja opóźnienia (odległość między
;próbkami)
;Czas opóźnienia zależy od komórki MAGSP ustawia-
;nej zleceniem *7.
;%
```

0702: 3A B2 FF
0705: 3D
0706: 20 FD
0708: C9

```
DEL02: LD  A,(MAGSP)
DE1:  DEC  A
      JR  NZ,DE1
      RET
```

;RESMAG - zerowanie wyjścia na magnetofon.

0709:
0709: AF
070A: 32 F4 FF
070D: D3 F0
070F: 3E 08
0711: D3 F3
0713: C9

```
RESMAG:
      XOR  A ;A=0
      LD  (KLA),A ;Płytki MIK94
      OUT (PA),A ;MIK94
      LD  A,8
      OUT (CONTR),A ;Płytki MIK90
      RET
```

```
;.COMMENT%
;M6 - odczyt z magnetofonu
;*[NAZWA][CR] - odczyt programu o nazwie [NAZWA]
;Jeśli nazwa odczytana z magnetofonu jest identy-
;czna jak [NAZWA] deklarowana w zleceniu *G
;to następuje wczytanie programu do pamięci
;typu RAM. Jeśli nazwy są różne to nastąpi jedynie
;wyswietlenie nazwy odczytanej z magnetofonu
;bez wpisywania programu do pamięci RAM.
;Powyższa właściwość umożliwia proste przeszuki-
;wanie zbiorów zapisanych na taśmie magnetofo-
```

```

;nowej - wystarczy zadeklarowac falszywa nazwe,
;aby w ciagu ok. 1 sek otrzymac nazwe zbioru
;zapisanego na magnetofonie.
;%
0714: 0D
0715: CD 13 02
0718: 20
0719: C1
071A: 41
M6:      DEC    C      ;1 Parametr
        CALL   EXPR    ;Pobranie nazwy
        DB     20H
        POP    BC
        LD     B,C     ;B - nazwa deklarowana
; .COMMENT%
;OMAG - procedura systemowa
;Odczyt programu o nazwie deklarowanej w rej. B.
;WEJ: B - nazwa deklarowana
;WYJ:
;1. Bezwarunkowe wyjście z procedury nastąpi
;   wraz z wcisnięciem klawisza "M"
;2. Wyjście nastąpi również po napotkaniu rekordu
;   EOF o nazwie zgodnej z zadeklarowaną. W tym
;   przypadku przewidziano dwa rodzaje wyjść:
;A. Jeśli procedura OMAG wywoływana jest z pro-
;gramu użytkownika to nastąpi skok do adresu
;wejścia podanego w rekordzie EOF (rozkaz typu JP)
;Na stosie przechowywany jest adr. powrotu do
;programu wywołującego (użytkownika). Jeśli zatem
;ADR. WEJ. wskazywał będzie rozkaz "RET" to nast-
;ąpi powrót do programu wywołującego.
;B. Jeśli procedura OMAG wywoływana jest z pro-
;gramu MONITORA (zlecenie *6), to nastąpi załado-
;wanie adresu wejścia deklarowanego w rekordzie
;EOF do licznika rozkazów użytkownika i powrót
;do programu MONITORA.
;ZMIENIA: AF,DE,HL,C      STOS: 11
;
;Budowa pojedynczego rekordu:
;MARK,NAZWA,DLUG,ADRES,-SUMN,BLOK DANYCH,-SUMD
;MARK (2 bajty) - wyróżnik początku
;NAZWA (1 bajt) - nazwa zbioru zapisanego na
;   magnetofonie.
;DLUG (1 bajt) - długość bloku danych.
;Jeśli DLUG=0 to rekord EOF.
;ADRES (2 bajty) - adres ładowania bloku danych
;lub jeśli rekord EOF to adres wejścia do
;odczytywanego programu.
;SUMN (1 bajt) - suma kontrolna nagłówka.
;<NAZWA,ADRES> (Rozkaz ADD).
;SUMD (1 bajt) - suma kontrolna bloku danych.
;%
071B:
071B: C5
        PUSH   BC      ;STOS - nazwa deklarowana
;Poszukiwanie wyróżnika "MARK"
RED1:   LD     HL,MARK
RED0:   CALL   RBYT     ;Rej. A - odczyt. bajt
REX:    CP     L        ;Porown. młodszych bajt.
        JR     NZ,RED0
        CALL   RBYT     ;Pobierz nast. bajt
        CP     H        ;Porow. starsz. bajtów
        JR     NZ,REX
;Znaleziono wyróżnik "MARK"
;Odczyt. parametrow: NAZWA,DLUG,ADRES,-SUMN

```


072B: 16 00	LD D,0 ;Zerowanie sumy kontrol.
072D: CD 79 07	CALL RBYT ;Odczyt. nazwy
0730: 5F	LD E,A ;E - nazwa z magnetofonu
0731: DF	RST LBYTE ;Wyswietl. nazwy
0732: 25	DB 25H
0733: CD 79 07	CALL RBYT ;Dlug. bloku danych
0736: 47	LD B,A ;B - dlug
0737: CD 79 07	CALL RBYT
073A: 6F	LD L,A ;Mlodszy bajt adresu
073B: CD 79 07	CALL RBYT
073E: 67	LD H,A ;Starszy bajt adresu
073F: E7	RST LADR ;Wysw. adresu
0740: 40	DB 40H
0741: CD 79 07	CALL RBYT ; -SUMN
0744: 20 22	JR NZ,ERRO ;Blad SUMN (CY=0)
0746: F1	POP AF
0747: F5	PUSH AF ;A - nazwa deklarowana
0748: BB	CP E ;Porownanie nazw
0749: 20 D1	JR NZ,RED1 ;Nazwy rozne
	;Odczyt naglowka bezbledny. rej.D=0
	;Sprawdzenie czy rekord EOF
074B: 78	LD A,B ;DLUG
074C: B7	OR A ;Czy DLUG=0 ?
074D: 28 19	JR Z,REOF ;DLUG=0 - rekord EOF
	;Wyswietlenie symbolu odczytywania "="
074F: 3E 48	LD A,ROWN ;Znak "="
0751: 32 FB FF	LD (BWYS+4),A ;Wyswietlenie "="
	;Rekord z blokiem danych.
	;Odczytanie bloku danych.
0754:	RED2:
0754: CD 79 07	CALL RBYT ;Pobierz dana
0757: 77	LD (HL),A ;Wpisanie do pamieci
0758: 23	INC HL
0759: 10 F9	DJNZ RED2
	;Koniec odczytywania bloku danych.
	;Sprawdzenie sumy kontrolnej bloku danych.
075B: CD 79 07	CALL RBYT ;Pobranie -SUMD
	;Kasowanie symbolu odczytywania "="
075E: 3E 00	LD A,ZGAS
0760: 32 FB FF	LD (BWYS+4),A ;Wygazenie symbolu
0763: 37	SCF ;CY=1 - blad SUMD
0764: 20 02	JR NZ,ERRO ;Blad SUMD (CY=1)
	;Blok danych odczytany w sposob bezbledny.
0766: 18 B4	JR RED1 ;Czytaj nast. rekord
	;Obsluga rekordu EOF
0768:	REOF:
0768: C1	ERRO: POP BC ;B - nazwa deklarowana
0769: C2 B8 FF	JP NZ,ERRMAG ;Proc. obslug. bledu
	;GSTAT=0 - wywołanie z programu MONITORA.
	;GSTAT#0 - wywołanie z programu uzytkownika
076C: 3A B3 FF	LD A,(GSTAT)
076F: B7	OR A
0770: 20 03	JR NZ,MONJES ;Program MONITORA
	;Wywołanie z programu uzytkownika
0772: D7	RST CLR ;Wygazenie wyswietlacza
0773: 80	DB 80H
0774: E9	JP (HL) ;Skok do prog. uzytkow.
	;Obsluga wywołania z progr. monitora
	;Wpisanie odczyt. z magnet. adresu wejscia

```

;do licznika rozkazow uzytkownika.
0775: MONJES:
0775: 22 A9 FF      LD      (PLOC-1),HL ;adr. PC - uzytkow.
0778: C9           RET      ;Powr. do progr. monitora
; .COMMENT%
;RBYT - procedura pomocnicza
;Odczytanie jednego bajtu z magnetofonu.
;Obliczanie sumy kontrolnej w rej. D.
;WEJ: -
;WYJ: A - odczytany bajt
;      D - aktualny stan sumy kontrolnej.
;      D:=D+ODCZYTANY BAJT (modulo256)
;ZMIENIA: AF,D,C      STOS: 6
;%
0779: E5      RBYT:  PUSH  HL
077A: D5              PUSH  DE
077B: C5              PUSH  BC      ;Ochrona BC,DE,HL
077C: CD 81 07      RBTX:  CALL  BSTAR
077F: 18 FB          JR      RBTX
;BSTAR - oczekiwanie na 2 bity stopu
0781: 0E 35      BSTAR:  LD      C,HIG2+4
0783: 0D          BST1:  DEC     C
0784: 28 0A          JR      Z,RBY ;Rozpoznano stop bit
0786: CD 02 07      CALL  DEL02 ;Opoznienie
0789: DB F0          IN      A,(PA)
078B: E6 80          AND     80H ;Wyizolow. bitu B7
078D: 28 F4          JR      Z,BST1 ;Odliczanie
078F: C9          RET      ;Nie stop bit
;RBY - odczytanie jednego bajtu
0790: 2E 80      RBY:    LD      L,80H
0792: 1E 00          LD      E,0
0794: CD D6 07      CALL  LICZ ;Oczek. na start bit
0797: 1C          INC     E ;E#0
0798: CD D6 07      CALL  LICZ ;Pobranie jedynki
079B: FE 1D          CP      HIG1
079D: D0          RET     NC
079E: FE 09          CP      LOW1
07A0: D8          RET     C
;LOW1=< A < HIG1 - rozpoznano start bit
07A1: 1D          DEC     E ;E=0
07A2: CD D6 07      RB1:  CALL  LICZ ;Pobranie probek
07A5: FE 1D          CP      HIG1
07A7: 30 1B          JR      NC,RB2 ;A>=HIG1
07A9: FE 09          CP      LOW1
07AB: D8          RET     C
;LOW1=< A < HIG1 - pojedyncze zero (1*0) lub
;      pojedyncza jedynka (1*1)
07AC: 7B          LD      A,E
07AD: 2F          CPL
07AE: 5F          LD      E,A
07AF: CD D6 07      CALL  LICZ ;Pobranie probek
07B2: FE 1D          CP      HIG1
07B4: D0          RET     NC
07B5: FE 09          CP      LOW1
07B7: D8          RET     C
;LOW1=< A < HIG1 - pojedyncza jedynka (1*1) lub
;      pojedyncze zero (1*0)
;1*0 i 1*1 to odczytany bit = 1
;1*1 i 1*0 to odczytany bit = 0

```

```

07B8: 7B
07B9: 1F
07BA: 7D
07BB: 1F
07BC: 6F
07BD: 38 0D
07BF: 7B
07C0: 2F
07C1: 5F
07C2: 18 DE
07C4: FE 31
07C6: D0
07C7: FE 1D
07C9: D8

07CA: 18 EC

07CC:
07CC: E1
07CD: C1
07CE: D1
07CF: E1

07D0: 4F
07D1: 82
07D2: 57
07D3: B7
07D4: 79
07D5: C9

07D6: 06 00
07D8: CD 02 07
07DB: 0C
07DC: 7B
07DD: B7
07DE: DB F0
07E0: 28 01
07E2: 2F
07E3: E6 80
07E5: 28 F1
07E7: 16 03
07E9: 04
07EA: 15
07EB: 79
07EC: 48
07ED: C8

RB3: LD A,E
RRA ;Ustawienie CY
LD A,L
RRA
LD L,A ;Zapam. pobr. bitu
JR C,KBYT ;Pob. wszystkie bity
LD A,E
CPL
LD E,A
JR RB1
RB2: CP HIG2
RET NC
CP LOW2
RET C
;LOW2=< A <HIG2 - podwojne zero (2*0) lub
; podwojna jedynka (2*1).
;2*0 to odczytany bit =0
;2*1 to odczytany bit =1
JR RB3
;Koniec procesu odczytywania pojedynczego bajtu
KBYT:
POP HL ;Kasow. powr. do BSTAR
POP BC
POP DE
POP HL ;Odtworzenie BC,DE,HL
;Obliczanie sumy kontrolnej
LD C,A ;Ochrona odczyt. bajtu
ADD A,D ;modulo256
LD D,A ;D - suma kontrolna
OR A ;CY=0
LD A,C ;Odtw. odczyt. bajtu
RET
;.COMMENT%
;LICZ - procedura pomocnicza.
;Zbieranie probek dopoty, dopoki nie napotkane
;zostana 3 kolejne probki przeciwnie.
;WEJ: E#0 - zliczanie jedynek
; E=0 - zliczanie zer
; C - ilosc probek juz zliczonych
;WYJ: A - probki pobrane
; C - probki przeciwnie
;ZMIENIA: AF,C,D STOS: 1
;%
LICZ: LD B,0
LICZ1: CALL DEL02 ;Opoznienie
INC C ;Licznik probek
LD A,E
OR A
IN A,(PA)
LIX: JR Z,LI0 ;Probki 0
CPL ;Probki 1
LI0: AND 80H ;Wyizolowanie bit7
JR Z,LICZ1
LD D,3 ;Max. 3 probki przeciwnie
LI1: INC B ;Licznik probek przeciwnie
DEC D
LD A,C ;A - probki pobrane
LD C,B ;C - probki przeciwnie
RET Z ;Wroc gdy D=0

```

```

07EE: 4F          LD      C,A
07EF: CD 02 07     CALL    DEL02
07F2: 7B          LD      A,E
07F3: B7          OR      A
07F4: DB F0        IN      A,(PA)
07F6: 28 01        JR      Z,LI2 ;Probki 0
07F8: 2F          CPL      ;Probki 1
07F9: E6 80        LI2:    AND     80H
07FB: 20 EC        JR      NZ,LI1
07FD: 0C          INC     C      ;Weryfikacja przeklamania
07FE: 18 D6        JR      LICZ

FF8D:            ORG     FF8DH
; .COMMENT%
; Obszar pamieci RAM angazowany przez program
; monitora systemu CA80.
; Obszar <FF97,FFD1> jest inicjowany po
; wlaczeniu zasilania !
; Obszar <FFC1,FFCC> jest inicjowany wraz
; z wcisnieciem klawisza "M".
; %
FF8D:            TOS:    ;Dno stosu systemowego
FF8D:            MTOP    EQU    TOS/256 ;Starszy bajt TOS
; .COMMENT%
; Obszar przechowywania rejestrow uzytkownika
; w czasie gdy wykonywany jest program monitora.
; %
FF8D: 00        ELOC:    DB      0      ;E
FF8E: 00        DLOC:    DB      0      ;D
FF8F: 00        CLOC:    DB      0      ;C
FF90: 00        BLOC:    DB      0      ;B
FF91: 00        FLOC:    DB      0      ;F
FF92: 00        ALOC:    DB      0      ;A
FF93: 00 00      DW      0      ;IX
FF95:            IXLOC    EQU    $-1    ;Wskaz. starszy bajt IX
FF95: 00 00      DW      0      ;IY
FF97:            IYLOC    EQU    $-1    ;Wskaz. starszy bajt IY
FF97: 66 FF      DW      TOS-27H      ;SP
FF99:            SLOC     EQU    $-1    ;Wskaz.starszy bajt SP
; .COMMENT%
; EXIT - procedura wejścia do programu uzytkow.
;
; WEJ: SP=TOS - rej. SP musi wskazywac TOS !
; WYJ: odtworzenie rejestrow uzytkownika a nast-
; epnie skok do programu uzytkownika.
; (rozkaz JP PCUZYT).
; %
FF99:            EXIT:
FF99: D1        POP     DE
FF9A: C1        POP     BC
FF9B: F1        POP     AF
FF9C: DD E1     POP     IX
FF9E: FD E1     POP     IY      ;Odtw. IY,IX,AF,BC,DE
FFA0: E1        POP     HL
FFA1: F9        LD      SP,HL ;Odtworzenie SP uzytk.
; Praca krokowa powoduje wstawienie OUT (0F4H),A
; w miejsce ponizszych NOP
FFA2: 00        KROK:    NOP
FFA3: 00        NOP

```

FFA4: 21 00 C1	LD HL,HLUZYT ;Odtw. HL uzytkownika
FFA7:	LLOC EQU \$-2 ;L
FFA7:	HLOC EQU \$-1 ;Rej. H uzytkownika
FFA7: FB	EI ;Odblokowanie przerwan
FFA8: C3 00 C0	JP PCUZYT ;Odtw. PC uzytkownika
FFAB:	PLOC EQU \$-1 ;Wsk. starszy bajt PC
	;Pulapki programowe - zlecenie *G.
FFAB: 00 00	TLOC: DW 0 ;Pulapka1
FFAD: 00	DB 0
FFAE: 00 00	DW 0 ;Pulapka2
FFB0: 00	DB 0
	;Parametry transmisji magnetofonowej
FFB1: 10	DLUG: DB 16 ;Dlug. bloku danych
FFB2: 25	MAGSP: DB 25H ;Szybkosc transmisji
	;.COMMENT%
	;Klucze programowe
	;GSTAT=0 - wykonywany program uzytkownika
	;GSTAT#0 - wykonywany program monitora
	;;
	;ZESTAT=0 - maskowanie obslugi zegara w NMI
	;ZESTAT#0 - zegar obslugiwany
	;%
FFB3: FF	GSTAT: DB 0FFH
FFB4: FF	ZESTAT: DB 0FFH
	;.COMMENT%
	;Skoki posrednie
	;M8 - obsluga zlecenia zdefiniowanego przez
	; uzytkownika (klawisz "8")
	;ERRMAG - obsluga blednego odczytu rekordu
	; z magnetofonu.
	;EM - emulator
	;RTS - po wlaczeniu zasilania nastepuje skok
	; do RTS/803 JESLI pa1=1 (uklad U7/8255).
	;%
FFB5: C3 00 08	M8: JP 800H ;Zobacz pkt 1.11 (MIK05)
FFB8: C3 87 04	ERRMAG: JP ERROR
FFBB: C3 06 08	EM: JP 806H
FFBE: C3 03 08	RTS: JP 803H
	;.COMMENT%
	;Systemowe skoki posrednie - inicjowane wraz
	;z wcisnieciem klawisza "M".
	;APWYS - wskazuje polozenie parametru
	; wyswietlacza PWYS.
	;CSTS - procedura systemowa
	;CI - procedura systemowa
	;AREST - skok do procedury RESTAR wykonywanej
	;po napotkaniu rozkazu RST 30H/F7 w programie
	;uzytkownika - prawidlowe przejście z progra-
	;mu uzytkownika do programu monitora.
	;%
FFC1: F6 FF	APWYS: DW PWYS
FFC3: C3 30 01	CSTS: JP CSTSM ;Procedura systemowa
FFC6: C3 84 01	CI: JP CIM ;Procedura systemowa
FFC9: C3 46 05	AREST: JP RESTAR
FFCC: C9	NMIU: RET ;Proced. NMI uzytkow.
FFCD: 00 00	DW 0 ;NMIU: JP NMIUZYT
	;Tablica przerwan uzytkownika
FFCF: C3 87 04	INTU: JP ERROR ;Skok do obslugi bledu
	;Ponizsze komorki nie sa inicjowane po wlacz-

```

;eniu zasilania
FFD2:      INTU0 EQU  $-2
FFD2: 00 00 INTU1: DW  0
FFD4: 00 00 INTU2: DW  0
FFD6: 00 00 INTU3: DW  0
FFD8: 00 00 INTU4: DW  0
FFDA: 00 00 INTU5: DW  0
FFDC: 00 00 INTU6: DW  0
FFDE: 00 00 INTU7: DW  0
FFE0: 00 00 00 00 00 00 00 00 REZ: DS 8 ;Rezerwa
; .COMMENT%
; Liczniki programowe
; LCI,SYG - liczniki dla potrzeb procedury CI -
; obsługiwane w NMI.
; TIME - licznik binarny modulo256 (licz. do tyłu)
; zmniejszany co 2 mS w procedurze NMI /przez-
; naczony dla potrzeb użytkownika.
; %
FFE8: 00 LCI: DB 0
FFE9: 00 SYG: DB 0 ;Sygnal wcis. klaw.
FFEA: 00 TIME: DB 0 ;Licznik modulo256
; Zegar czasu rzeczywistego
; Odliczanie czasu w kodzie BCD
FFEB: 00 MSEK: DB 0 ;<0,4>
FFEC: 00 SETSEK: DB 0 ;<0,99> setne sek.
FFED: 00 SEK: DB 0 ;<0,59> sekundy
FFEE: 00 MIN: DB 0 ;<0,59> minuty
FFEF: 00 GODZ: DB 0 ;<0,23> godziny
FFF0: 00 DNITYG: DB 0 ;<7,6,5,4,3,2,1>
; Dni tygodnia - odliczanie do tyłu !!
FFF1: 00 DNIM: DB 0 ;<1...>dni miesiaca
FFF2: 00 MIES: DB 0 ;<1,12> miesiace
FFF3: 00 LATA: DB 0 ;<0,99> rok
; .COMMENT%
; KLAW - aktualny stan portu wyjsciowego PA/F0
; na plytce MIK94. Wykorzystuja procedury
; CSTS oraz NMI.
; SBUF - bity B7,B6,B5 wskazuja aktualnie wyswiet-
; lana cyfre z bufora BWYS. Wykorzystuje
; wylacznie procedura NMI.
; %
FFF4: 00 KLAW: DB 0
FFF5: 00 SBUF: DB 0
; Wswietlacz siedmiosegmentowy
FFF6: 00 PWYS: DB 0 ;Parametr wswietlacza
; Bufor wswietlacza siedmiosegmentowego
FFF7: BWYS:
FFF7: 00 CYF0: DB 0 ;Cyfra nr.0
FFF8: 00 CYF1: DB 0 ;Cyfra nr.1
FFF9: 00 CYF2: DB 0 ;Cyfra nr.2
FFFA: 00 CYF3: DB 0 ;Cyfra nr.3
FFFB: 00 CYF4: DB 0 ;Cyfra nr.4
FFFC: 00 CYF5: DB 0 ;Cyfra nr.5
FFFD: 00 CYF6: DB 0 ;Cyfra nr.6
FFFE: 00 CYF7: DB 0 ;Cyfra nr.7

; END CA80

```